Under consideration for publication in J. Functional Programming

A Type and Scope Safe Universe of Syntaxes with Binding: Their Semantics and Proofs

GUILLAUME ALLAIS, ROBERT ATKEY
University of Strathclyde (UK)
JAMES CHAPMAN
Input Output HK Ltd. (HK)
CONOR MCBRIDE
University of Strathclyde (UK)
JAMES MCKINNA
University of Edinburgh (UK)

Abstract

Almost every programming language's syntax includes a notion of binder and corresponding bound occurrences, along with the accompanying notions of α-equivalence, capture avoiding substitution, typing contexts, runtime environments, and so on. In the past, implementing and reasoning about programming languages required careful handling to maintain the correct behaviour of bound variables. Modern programming languages include features that enable constraints like scope safety to be expressed in types. Nevertheless, the programmer is still forced to write the same boilerplate over again for each new implementation of a scope safe operation (e.g., renaming, substitution, desugaring, printing, etc.), and then again for correctness proofs.

We present an expressive universe of syntaxes with binding and demonstrate how to (1) implement scope safe traversals once and for all by generic programming; and (2) how to derive properties of these traversals by generic proving. Our universe description, generic traversals and proofs, and our examples have all been formalised in Agda and are available in the accompanying material.

1 Introduction

In modern typed programming languages, programmers writing embedded DSLs (Hudak (1996)) and researchers formalising them can now use the host language's type system to help them. Using Generalised Algebraic Data Types (GADTs) or the more general indexed families of Type Theory (Dybjer (1994)) to represent syntax, programmers can *statically* enforce some of the invariants in their languages. For example, managing variable scope is a popular use case in LEGO, Idris, Coq, Agda and Haskell (Altenkirch & Reus (1999); Brady & Hammond (2006); Hirschowitz & Maggesi (2012); Keuchel & Jeuring (2012); Bach Poulsen *et al.* (2018); Wadler & Kokke (2018); Eisenberg (2018)) as directly manipulating raw de Bruijn indices is notoriously error-prone. Solutions have been proposed that range from enforcing well scopedness of variables to ensuring full type correctness. In short, these techniques use the host languages' types to ensure that "illegal states are unrepresentable”, where illegal states correspond to ill scoped or ill typed terms in the object language.

Despite the large body of knowledge in how to use types to define well formed syntax (see the Related Work in Section 10), it is still necessary for the working DSL designer or formaliser to redefine essential functions like renaming and substitution for each new syntax, and then to reprove essential lemmas about those functions. To reduce the burden of such repeated work and boilerplate, in this paper we apply the methodology of datatype-genericity to programming and proving with syntaxes with binding.

To motivate our approach, let us look at the formalisation of an apparently straightforward program transformation: the inlining of let-bound variables by substitution together with a soundness lemma proving that reductions in the source languages can be simulated by reductions in the target one. There are two languages: the source (S), which has let-bindings, and the target (T), which only differs in that it does not:

$$\mathsf{S} ::= x \mid \mathsf{S}\ \mathsf{S} \mid \lambda x.\mathsf{S} \mid \mathrm{let}\ x = \mathsf{S}\ \mathrm{in}\ \mathsf{S} \qquad \mathsf{T} ::= x \mid \mathsf{T}\ \mathsf{T} \mid \lambda x.\mathsf{T}$$

Breaking the task down, an implementer needs to define an operational semantics for each language, define the program transformation itself, and prove a correctness lemma that states each step in the source language is simulated by zero or more steps of the transformed terms in the target language. In the course of doing this, they will discover that there is actually a large amount of work:

1. To define the operational semantics, one needs to define substitution, and hence renaming. This needs to be done separately for both the source and target languages, even though they are very similar;
2. In the course of proving the correctness lemma, one needs to prove eight lemmas about the interactions of renaming, substitution, and transformation that are all remarkably similar, but must be stated and proved separately (e.g, as observed by (Benton *et al.* (2012))).

Even after doing all of this work, you have only a result for a single pair of source and target languages. If you were to change your languages S or T, you would have to repeat the same work all over again (or at least do a lot of cutting, pasting, and editing).

The main contribution of this paper is this: using the universe of syntaxes with binding we present in this paper, we are able to solve this repetition problem *once and for all*.

Content and Contributions. To introduce the basic ideas that this paper builds on, we start with primers on scoped and sorted terms (Section 2), scope and sort safe programs acting on them (Section 3), and programmable descriptions of data types (Section 4). These introductory sections help us build an understanding of the problem at hand as well as a toolkit that leads us to the novel content of this paper: a universe of scope safe syntaxes with binding (Section 5) together with a notion of scope safe semantics for these syntaxes (Section 6). This gives us the opportunity to write generic implementations of renaming and substitution (Section 6.2), a generic let-binding removal transformation (generalising the problem stated above) (Section 7.5), and normalisation by evaluation (Section 7.7). Further, we show how to construct generic proofs by formally describing what it means for one semantics to simulate another (Section 9.1), or for two semantics to be fusable

(Section 9.2). This allows us to prove the lemmas required above for renaming, substitution, and desugaring of let binders generically, for *every* syntax in our universe.

Our implementation language is Agda (Norell (2009)). However, our techniques are language independent: any dependently typed language at least as powerful as Martin-Löf Type Theory (Martin-Löf (1982)) equipped with inductive families (Dybjer (1994)) such as Coq (The Coq Development Team (2017)), Lean (de Moura *et al.* (2015)) or Idris (Brady (2013)) ought to do.

Changes with respect to the ICFP 2018 version This paper is a revised and expanded version of a paper of the same title that appeared at ICFP 2018. This extended version of the paper includes many more examples of the use of our universe of syntax with binding for writing generic programs in Section 7: pretty printing with human readable names (Section 7.1), scope checking (Section 7.2), type checking (Section 7.3), elaboration (Section 7.4), inlining of single use let-bound expressions (shrinking reductions) (Section 7.6), and normalisation by evaluation (Section 7.7). We have also included a discussion of how to define generic programs for deciding equality of terms. Additionally, we have elaborated our descriptions and examples throughout, and expanded our discussion of related work in Section 10.

2 A Primer on Scope And Sort Safe Terms

From Inductive Types to Inductive Families for Abstract Syntax A reasonable way to represent the abstract syntax of the untyped λ-calculus in a typed functional programming language is to use an inductive type:

```
data Lam : Set where
  'var  :  ℕ → Lam
  'lam  :  Lam → Lam
  'app  :  Lam → Lam → Lam
```

We have used de Bruijn (1972) indices to represent variables by the number of 'lam binders one has to pass up through to reach the binding occurrence. The de Bruijn representation has the advantage that terms are automatically represented up to α-equivalence. If the index goes beyond the number of binders enclosing it, then we assume that it is referring to some context, left implicit in this representation.

This representation works well enough for writing programs, but the programmer must constantly be vigilant to guard against the accidental construction of ill-scoped terms. The implicit context that accompanies each represented term is prone to being forgotten or muddled with another, leading to confusing behaviour when variables either have dangling pointers or point to the wrong thing.

To improve on this situation, previous authors have proposed to use the host language's type system to make the implicit context explicit, and to enforce well scopedness of variables. Scope safe terms follow the discipline that every variable is either bound by some binder or is explicitly accounted for in a context. Bellegarde and Hook (1994), Bird and Patterson (1999), and Altenkirch and Reus (1999) introduced the classic presentation of scope safety using inductive *families* (Dybjer (1994)) instead of plain inductive types to represent abstract syntax. Indeed, using a family indexed by a **Set**, we can track scoping information at the type level. The empty **Set** represents the empty scope. The type constructor $1 + (\_)$ extends the running scope with an extra variable.

```
data Lam : Set → Set where
  ‘var  :  X → Lam X
  ‘lam  :  Lam (1+X) → Lam X
  ‘app  :  Lam X → Lam X → Lam X
```

The *Lam* type is now a family of types, indexed by the set of variables in scope. Thus, the context for each represented term has been made visible to the type system, and the types enforce that only variables that have been explicitly declared can be referenced in the ‘var constructor. We have made illegal terms unrepresentable.

Since ‘Lam’ is defined to be a function Set → Set, it makes sense to ask whether it is also a functor and a monad. Indeed it is, as Altenkirch and Reus showed. The functorial action corresponds to renaming, the monadic ‘return’ corresponds to the use of variables (the ‘var constructor), and the monadic ‘bind’ corresponds to substitution. The functor and monad laws correspond to well known properties from the equational theories of renaming and substitution. We will revisit these properties, for our whole universe of syntax with binding, in Section 9.2.

A Mechanized and Typed Variant of Altenkirch and Reus’ Calculus. There is no reason to restrict this technique to inductive families indexed by **Set**. The more general case of inductive families in $\mathbf{Set}^J$ can be endowed with similar functorial and monadic operations by using Altenkirch, Chapman and Uustalu’s relative monads (2015; 2014).

We pick as our index type J the category whose objects are inhabitants of List I (I is a parameter of the construction) and whose morphisms are thinnings (permutations that may forget elements, see Section 3.1). Values of type List I are intended to represent the list of the sorts (or kinds, or types, depending on the application) of the de Bruijn variables in scope. We can recover an unsorted approach by picking I to be the unit type. Given this sorted setting, our functors take an extra I argument corresponding to the sort of the expression being built. This is captured by the large type I –Scoped:

```
_–Scoped : Set → Set₁
I –Scoped = I → List I → Set
```

We use Agda’s mixfix operator notation where underscores denote argument positions.

To lighten the presentation, we exploit the observation that the current scope is either passed unchanged to subterms (e.g. in the application case) or extended (e.g. in the λ-abstraction case) by introducing combinators to build indexed types, conforming to the convention (see e.g. (Martin-Löf (1982))) of mentioning only context *extensions* when presenting judgements. Indeed, this observation is critical to our whole approach to syntax with binding, and will arise again in our generic formulation of syntax traversals in Section 6.

```
_⇒_ : (P Q : A → Set) → (A → Set)        _⊢_ : (A → B) → (B → Set) → (A → Set)
(P ⇒ Q) x = P x → Q x                     (f ⊢ P) x = P (f x)

const : Set → (A → Set)                   ∀[_] : (A → Set) → Set
const P x = P                             ∀[_] P = ∀ {x} → P x
```

We lift the function space pointwise with _⇒_, silently threading the underlying scope. The _⊢_ makes explicit the *adjustment* made to the index by a function. We write $f \vdash T$ where f is the modification and T the indexed Set it operates on. Although it may seem surprising at first to define binary infix operators as having arity three, they are meant to be used partially applied, surrounded by ∀[_] which turns an indexed Set into a Set by implicitly quantifying over the index. Lastly, const is the constant combinator, which ignores the index.

We make ⇒ associate to the right as one would expect and give it the highest precedence level as it is the most used combinator. These combinators lead to more readable type declarations. For instance, the compact expression ∀[s ⊢ (const P ⇒ Q) ⇒ R] desugars to the more verbose type ∀ {i} → (P → Q (s i)) → R i.

As the context argument comes second in the definition of _–Scoped, we can readily use these combinators to thread, modify, or quantify over the scope when defining such families:

```
data Var : I –Scoped where                data Lam : Type –Scoped where
  z : ∀[ (σ ::_) ⊢ Var σ ]                  'var  : ∀[ Var σ ⇒ Lam σ ]
  s : ∀[ Var σ ⇒ (τ ::_) ⊢ Var σ ]          'app : ∀[ Lam (σ '→ τ) ⇒ Lam σ ⇒ Lam τ ]
                                            'lam  : ∀[ (σ ::_) ⊢ Lam τ ⇒ Lam (σ '→ τ) ]
```

The inductive family Var represents well scoped and well kinded de Bruijn indices. Its z (for zero) constructor refers to the nearest binder in a non-empty scope. The s (for successor) constructor lifts a a variable in a given scope to the extended scope where an extra variable has been bound. Both of the constructors' types have been written using the combinators defined above. They respectively normalise to:

```
z : ∀ σ Γ → Var σ (σ :: Γ)        s : ∀ σ τ Γ → Var σ Γ → Var σ (τ :: Γ)
```

We will reuse the Var family to represent variables in all the syntaxes defined in this paper.

The Type –Scoped family Lam is Altenkirch and Reus' intrinsically typed representation of the simply typed λ-calculus, where Type is the Agda type of simple types. We can readily write well scoped-and-typed terms such as e.g. application, a closed term of type ((σ '→ τ) '→ (σ '→ τ)) ({- and -} delimit comments meant to help the reader see which binder the de Bruijn indices are referring too):

```
apply : Lam ((σ '→ τ) '→ (σ '→ τ)) []
apply = 'lam {- f -} $ 'lam {- x -}
        $ 'app ('var (s z) {- f -}) ('var z {- x -})
```

3 A Primer on Type and Scope Safe Programs

The scope- and type- safe representation described in the previous section is naturally only a start. Once the programmer has access to a good representation of the language they are interested in, they will want to write programs manipulating terms. Renaming and substitution are the two typical examples that are required for almost all syntaxes. Now that well-typedness and well-scopedness are enforced statically, all of these traversals have to be implemented in a type and scope safe manner. These constraints show up in the types of renaming and substitution:

```
ren : (Γ −Env) Var Δ → Lam σ Γ → Lam σ Δ
ren ρ (‘var k)   = var_r (lookup ρ k)
ren ρ (‘app f t) = ‘app (ren ρ f) (ren ρ t)
ren ρ (‘lam b)   = ‘lam (ren (extend_r ρ) b)

sub : (Γ −Env) Lam Δ → Lam σ Γ → Lam σ Δ
sub ρ (‘var k)   = var_s (lookup ρ k)
sub ρ (‘app f t) = ‘app (sub ρ f) (sub ρ t)
sub ρ (‘lam b)   = ‘lam (sub (extend_s ρ) b)
```

Fig. 1. Type and Scope Preserving Renaming and Substitution

We have intentionally hidden technical details behind some auxiliary definitions left abstract here: var and extend. Their implementations are distinct for ren and sub but they serve the same purpose: var is used to turn a value looked up in the evaluation environment into a term and extend is used to alter the environment when going under a binder. This presentation highlights the common structure between ren and sub which we will exploit later in this section, particularly in Section 3.2 where we define an abstract notion of semantics and the corresponding generic traversal.

3.1 A Generic Notion of Environments

Both renaming and substitution are defined in terms of *environments* $(\Gamma\ {-}\mathsf{Env})\ \mathcal{V}\ \Delta$ that describe how to associate a value $\mathcal{V}$ (variables for renaming, terms for substitution) well scoped and typed in Δ to every entry in Γ. Environments are defined as the following record structure (using a record helps Agda's type inference reconstruct the type family $\mathcal{V}$ of values for us):

```
record _−Env (Γ : List I) (𝒱 : I −Scoped) (Δ : List I) : Set where
  constructor pack
  field lookup : Var i Γ → 𝒱 i Δ
```

As we have already observed, the definitions of renaming and substitution have very similar structure. Abstracting away this shared structure would allow for these definitions to be refactored, and their common properties to be proved in one swift move.

Previous efforts in dependently typed programming (Benton *et al.* (2012); Allais *et al.* (2017)) have achieved this goal and refactored renaming and substitution, but also normalisation by evaluation, printing with names or CPS conversion as various instances of a more general traversal. As we will show in Section 7.3, typechecking in the style of Atkey (2015) also fits in that framework. To make sense of this body of work, we need to introduce three new notions: Thinning, a generalisation of renaming; Thinnables which are types that permit thinning; and the □ functor, which freely adds Thinnability to any indexed type. We use □, and our compact notation for the indexed function space between indexed types, to crisply encapsulate the additional quantification over environment extensions which is typical of Kripke semantics.

```
Thinning : List I → List I → Set
Thinning Γ Δ = (Γ −Env) Var Δ
```

The Special Case of Thinnings Thinnings subsume more structured notions such as the Category of Weakenings (Altenkirch *et al.* (1995)) or Order Preserving Embeddings (Chapman (2009)), cf. Figure 2 for some examples of combinators. In particular, they do not prevent the user from defining arbitrary permutations or from introducing contractions although we will not use such instances. However, such extra flexibility will not get in our way, and permits a representation as a function space which grants us monoid laws "for free" as per Jeffrey's observation (2011).

```
identity : Thinning Γ Γ
lookup identity k = k

extend : Thinning Γ (σ :: Γ)
lookup extend v = s v

select : Thinning Γ Δ → (Δ −Env) 𝒱 Θ → (Γ −Env) 𝒱 Θ
lookup (select ren ρ) k = lookup ρ (lookup ren k)
```

Fig. 2. Identity Thinning, context extension, and (generalised) transitivity

The □ combinator turns any (List *I*)-indexed Set into one that can absorb thinnings. This is accomplished by abstracting over all possible thinnings from the current scope, akin to an S4-style necessity modality. The axioms of S4 modal logic incite us to observe that the functor □ is a comonad: extract applies the identity Thinning to its argument, and duplicate is obtained by composing the two Thinnings we are given. The expected laws hold trivially thanks to Jeffrey's trick mentioned above.

The notion of Thinnable is the property of being stable under thinnings; in other words Thinnables are the coalgebras of □. It is a crucial property for values to have if one wants to be able to push them under binders. From the comonadic structure we get that the □ combinator freely turns any (List I)-indexed Set into a Thinnable one.

```
□ : (List I → Set) → (List I → Set)
(□ T) Γ = ∀[ Thinning Γ ⇒ T ]

Thinnable : (List I → Set) → Set
Thinnable T = ∀[ T ⇒ □ T ]

extract : ∀[ □ T ⇒ T ]
extract t = t identity

duplicate : ∀[ □ T ⇒ □ (□ T) ]
duplicate t ρ σ = t (select ρ σ)

th^□ : Thinnable (□ T)
th^□ = duplicate
```

Fig. 3. The □ comonad, Thinnable, and the cofree Thinnable.

3.2 A Generic Notion of Semantics

As Allais, Chapman, McBride and McKinna (ACMM) (2017) show, equipped with these new notions we can define an abstract concept of semantics for our scope- and type- safe language. Provided that a set of constraints on two (Type −Scoped) $\mathcal{V}$ and C is satisfied, we will obtain a traversal of the following type:

```
semantics : (Γ −Env) 𝒱 Δ → (Lam σ Γ → C σ Δ)
```

Broadly speaking, a semantics turns our deeply embedded abstract syntax trees into the shallow embedding of the corresponding parametrised higher order abstract syntax term. We get a choice of useful scope- and type- safe traversals by using different 'host languages' for this shallow embedding.

Semantics, specified in terms of a record Semantics, are defined in terms of a choice of values $\mathcal{V}$ and computations C. A semantics must satisfy constraints on the notions of values $\mathcal{V}$ and computations C at hand.

```
record Semantics (𝒱 C : Type −Scoped) : Set where
```

First of all, values $\mathcal{V}$ should be Thinnable so that semantics may push the environment under binders. This constraint allows us to define extend which takes a context extension from Δ to Θ in the form of a thinning, an existing evaluation environment mapping Γ variables to Δ values and a value living in the extended context Θ and returns an evaluation environment mapping (σ :: Γ) variables to Θ values.

```
th^𝒱 : Thinnable (𝒱 σ)

extend : Thinning Δ Θ → (Γ −Env) 𝒱 Δ → 𝒱 σ Θ → (σ :: Γ −Env) 𝒱 Θ
extend σ ρ v = (λ t → th^𝒱 t σ) <$> ρ • v
```

Second, the set of computations needs to be closed under various combinators which are the semantical counterparts of the language's constructors. For instance in the variable case we obtain a value from the evaluation environment but we need to return a computation. This means that values should embed into computations.

```
var : ∀[ 𝒱 σ ⇒ C σ ]
```

The semantical counterpart of application is an operation that takes a representation of a function and a representation of an argument and produces a representation of the result.

```
app : ∀[ C (σ '→ τ) ⇒ C σ ⇒ C τ ]
```

The interpretation of the λ-abstraction is of particular interest: it is a variant on the Kripke function space one can find in normalisation by evaluation. In all possible thinnings of the scope at hand, it promises to deliver a computation whenever it is provided with a value for its newly bound variable. This is concisely expressed by the constraint's type:

```
lam : ∀[ □ (𝒱 σ ⇒ C τ) ⇒ C (σ '→ τ) ]
```

Agda allows us to package, together with the fields of the record Semantics, the generic traversal function semantics, which is brought into scope for any instance of Semantics.

We thus realise the promise made earlier, namely that any given Semantics $\mathcal{V}$ C induces a function which, given a value in $\mathcal{V}$ for each variable in scope, transforms a Lam term into a computation C.

```
semantics : (Γ –Env) 𝒱 Δ → (Lam σ Γ → C σ Δ)
semantics ρ (‘var k)   = var (lookup ρ k)
semantics ρ (‘app f t) = app (semantics ρ f) (semantics ρ t)
semantics ρ (‘lam b)   = lam (λ σ v → semantics (extend σ ρ v) b)
```

Fig. 4. Fundamental Lemma of Semantics for Lam, relative to a given Semantics $\mathcal{V}$ C

3.3 Instances of Semantics

Coming back to renaming and substitution, we see that they both fit in the Semantics framework. We notice that the definition of substitution depends on the definition of renaming: to be able to push terms under binder, we need to have already proven that they are thinnable.

```
Renaming : Semantics Var Lam
Renaming = record
  { th^𝒱 = th^Var
  ; var  = ‘var
  ; app  = ‘app
  ; lam  = λ b → ‘lam (b extend z) }

ren : (Γ –Env) Var Δ → Lam σ Γ → Lam σ Δ
ren = Semantics.semantics Renaming
```

```
Substitution : Semantics Lam Lam
Substitution = record
  { th^𝒱 = λ t ρ → ren ρ t
  ; var  = id
  ; app  = ‘app
  ; lam  = λ b → ‘lam (b extend (‘var z)) }

sub : (Γ –Env) Lam Δ → Lam σ Γ → Lam σ Δ
sub = Semantics.semantics Substitution
```

Fig. 5. Renaming and Substitution as Instances of Semantics

In both cases we use (extend) defined in Figure 2 as (pack s) (where pack is the constructor for environments and s, defined in Section 2, is the function lifting an existing de Bruijn variable into an extended scope) as the definition of the thinning embedding Γ into σ :: Γ.

We also include the definition of a basic printer relying on a name supply to highlight the fact that computations can very well be effectful. The Printing semantics is defined by using Names (i.e. Strings) as values and Printers (i.e. State (Stream String) String) as computations. We use a Wrapper with a type and a context as phantom types in order to help Agda's inference propagate the appropriate constraints. We define a function fresh that generates new concrete names using a State monad.

The wrapper Wrap does not depend on the scope Γ so it is automatically a Thinnable functor. We jump straight to the definition of the printer. To print an application, we produce a string representation of the term in function position, then one of its argument and combine them by putting the argument between parentheses. To print a λ-abstraction, we start by generating a fresh name for the newly-bound variable, use that name to generate a string representing the body of the function to which we prepend a "λ" binding the fresh name.

```
record Wrap (A : Set) (σ : I) (Γ : List I) : Set where        M : Set → Set
  constructor MkW; field getW : A                             M = State (Stream String _)

Name : I –Scoped                       fresh : ∀ σ → M (Name σ (σ :: Γ))
Name = Wrap String                     fresh σ = do
                                         names ← get
Printer : I –Scoped                      put (tail names)
Printer = Wrap (M String)                pure (MkW (head names))
```

Fig. 6. Wrapper and fresh name generation

```
Printing : Semantics Name Printer
Printing = record
  { th^𝒱 = th^Wrap
  ; var  = map^Wrap return
  ; app  = λ mf mt → MkW $ getW mf >>= λ f → getW mt >>= λ t →
           return $ f ++ " (" ++ t ++ ")"
  ; lam  = λ {σ} mb → MkW $ fresh σ >>= λ x →
           getW (mb extend x) >>= λ b →
           return $ "λ" ++ getW x ++ ". " ++ b }
```

Fig. 7. Printing as an instance of Semantics

We can readily run the printer on some examples. Using apply defined in Figure 2, we get:

```
_ : print _ apply ≡ "λa. λb. a (b)"
_ = refl
```

Both printing and renaming highlight the importance of distinguishing values and computations: the type of values in their respective environments are distinct from their type of computations.

All of these examples are already described at length by ACMM (2017) so we will not spend any more time on them. They have also obtained the simulation and fusion theorems demonstrating that these traversals are well behaved as corollaries of more general results expressed in terms of semantics. We will come back to this in Section 9.1.

One important observation to make is the tight connection between the constraints described in Semantics and the definition of Lam: the semantical counterparts of the Lam constructors are obtained by replacing the recursive occurrences of the inductive family with either a computation or a Kripke function space depending on whether an extra variable was bound. This suggests that it ought to be possible to compute the definition of Semantics from the syntax description. Before doing this in Section 5, we need to look at a generic descriptions of datatypes.

4 A Primer on Universes of Data Types

Chapman, Dagand, McBride and Morris (CDMM) (2010) defined a universe of data types inspired by Dybjer and Setzer's finite axiomatisation of Inductive-Recursive definitions (1999) and Benke, Dybjer and Jansson's universes for generic programs and proofs (2003). This explicit definition of *codes* for data types empowers the user to write generic programs tackling *all* of the data types one can obtain this way. In this section we recall the main aspects of this construction we are interested in to build up our generic representation of syntaxes with binding.

The first component of CDMM's universe's definition is an inductive type of Descriptions of strictly positive functors from $\mathbf{Set}^J$ to $\mathbf{Set}^I$. It has three constructors: ‘σ to store data (the rest of the description can depend upon this stored value), ‘X to attach a recursive substructure indexed by J and ‘■ to stop with a particular index value.

The recursive function ⟦_⟧ makes the interpretation of the descriptions formal. Interpretation of descriptions give rise right-nested tuples terminated by equality constraints.

```
data Desc (I J : Set) : Set₁ where
  ‘σ : (A : Set) → (A → Desc I J) → Desc I J
  ‘X : J → Desc I J → Desc I J
  ‘■ : I → Desc I J
```

```
⟦_⟧ : Desc I J → (J → Set) → (I → Set)
⟦ ‘σ A d ⟧ X i = Σ[ a ∈ A ] (⟦ d a ⟧ X i)
⟦ ‘X j d  ⟧ X i = X j × ⟦ d ⟧ X i
⟦ ‘■ i′   ⟧ X i = i ≡ i′
```

Fig. 8. Datatype Descriptions and their Meaning as Functors

These constructors give the programmer the ability to build up the data types they are used to. For instance, the functor corresponding to lists of elements in A stores a Boolean which stands for whether the current node is the empty list or not. Depending on its value, the rest of the description is either the "stop" token or a pair of an element in A and a recursive substructure i.e. the tail of the list. The List type is unindexed, we represent the lack of an index with the unit type ⊤ whose unique inhabitant is tt.

```
listD : Set → Desc ⊤ ⊤
listD A = ‘σ Bool $ λ isNil →
            if isNil then ‘■ tt
            else ‘σ A (λ _ → ‘X tt (‘■ tt))
```

Fig. 9. The Description of the base functor for List A

Indexes can be used to enforce invariants. For example, the type Vec A n of length-indexed lists. It has the same structure as the definition of listD. We start with a Boolean distinguishing the two constructors: either the empty list (in which case the branch's index is enforced to be 0) or a non-empty one in which case we store a natural number n, the head of type A and a tail of size n (and the branch's index is enforced to be suc n).

```
vecD : Set → Desc ℕ ℕ
vecD A = ‘σ Bool $ λ isNil →
           if isNil then ‘■ 0
           else ‘σ ℕ (λ n → ‘σ A (λ _ → ‘X n (‘■ (suc n))))
```

Fig. 10. The Description of the base functor for Vec A n

The payoff for encoding our datatypes as descriptions is that we can define generic programs for whole classes of data types. The decoding function ⟦_⟧ acted on the objects of $\mathbf{Set}^J$, and we will now define the function fmap by recursion over a code d. It describes the action of the functor corresponding to d over morphisms in $\mathbf{Set}^J$. This is the first example of generic programming over all the functors one can obtain as the meaning of a description.

```
fmap : (d : Desc I J) → ∀[ X ⇒ Y ] → ∀[ ⟦ d ⟧ X ⇒ ⟦ d ⟧ Y ]
fmap (‘σ A d) f (a , v) = (a , fmap (d a) f v)
fmap (‘X j d)  f (r , v) = (f r , fmap d f v)
fmap (‘■ i)    f t       = t
```

Fig. 11. Action on Morphisms of the Functor corresponding to a Description

All the functors obtained as meanings of Descriptions are strictly positive. So we can build the least fixpoint of the ones that are endofunctors (i.e. the ones for which I equals J). This fixpoint is called μ and its iterator is given by the definition of fold d[1] .

```
data μ (d : Desc I I) : Size → I → Set where
  ‘con : ⟦ d ⟧ (μ d s) i → μ d (↑ s) i

fold : (d : Desc I I) → ∀[ ⟦ d ⟧ X ⇒ X ] → ∀[ μ d s ⇒ X ]
fold d alg (‘con t) = alg (fmap d (fold d alg) t)
```

Fig. 12. Least Fixpoint of an Endofunctor and Corresponding Generic Fold

We can see in Figure 13 that we can recover the types we are used to thanks to this least fixpoint, that pattern synonyms let us hide away the encoding and have the constructors we want, and that we can get our hands on the types' eliminators by instantiating fold.

```
List : Set → Set
List A = μ (listD A) ∞ tt

pattern [] = ‘con (true , refl)
pattern _::_ x xs = ‘con (false , x , xs , refl)
```

```
foldr : (A → B → B) → B → List A → B
foldr c n = fold (listD _) $ λ where
  (true             , refl) → n
  (false , hd , rec , refl) → c hd rec
```

Fig. 13. List, its constructors, and eliminator

The CDMM approach therefore allows us to generically define iteration principles for all data types that can be described. These are exactly the features we desire for a universe of data types with binding, so in the next section we will see how to extend CDMM's approach to include binding.

[1] **NB** In Figure 12 the Size (Abel (2010)) index added to the inductive definition of μ plays a crucial role in getting the termination checker to see that fold is a total function.

The functor underlying any well scoped and sorted syntax can be coded as some Desc (I × List I) (I × List I), with the free monad construction from CDMM uniformly adding the variable case. Whilst a good start, Desc treats its index types as unstructured, so this construction is blind to what makes the List I index a *scope*. The resulting 'bind' operator demands a function which maps variables in *any* sort and scope to terms in the *same* sort and scope. However, the behaviour we need is to preserve sort while mapping between specific source and target scopes which may differ. We need to account for the fact that scopes change only by extension, and hence that our specifically scoped operations can be pushed under binders by weakening.

5 A Universe of Scope Safe and Well Kinded Syntaxes

Our universe of scope safe and well kinded syntaxes follows the same principle as CDMM's universe of datatypes, except that we are not building endofunctors on Set any more but rather on *I* −Scoped. We now think of the index type *I* as the sorts used to distinguish terms in our embedded language. The 'σ and '■ constructors are as in the CDMM Desc type, and are used to represent data and index constraints respectively. What distinguishes this new universe Desc from that of Section 4 is that the 'X constructor is now augmented with an additional List *I* argument that describes the new binders that are brought into scope at this recursive position. This list of the kinds of the newly-bound variables will play a crucial role when defining the description's semantics as a binding structure in Figures 15, 16 and 17.

```
data Desc (I : Set) : Set₁ where
  'σ : (A : Set) → (A → Desc I) → Desc I
  'X : List I → I → Desc I        → Desc I
  '■ : I                          → Desc I
```

Fig. 14. Syntax Descriptions

The meaning function ⟦_⟧ we associate to a description follows closely its CDMM equivalent. It only departs from it in the 'X case and the fact it is not an endofunctor on *I* −Scoped; it is more general than that. The function takes an *X* of type List *I* → *I* −Scoped to interpret 'X Δ *j* (i.e. substructures of sort *j* with newly-bound variables in Δ) in an ambient scope Γ as *X* Δ *j* Γ.

```
⟦_⟧ : Desc I → (List I → I −Scoped) → I −Scoped
⟦ 'σ A d   ⟧ X i Γ = Σ[ a ∈ A ] (⟦ d a ⟧ X i Γ)
⟦ 'X Δ j d ⟧ X i Γ = X Δ j Γ × ⟦ d ⟧ X i Γ
⟦ '■ j     ⟧ X i Γ = i ≡ j
```

Fig. 15. Descriptions' Meanings

The astute reader may have noticed that ⟦_⟧ is uniform in *X* and Γ; however refactoring ⟦_⟧ to use the partially applied *X* _ _ Γ following this observation would lead to a definition harder to use with the combinators for indexed sets described in Section 2 which make our types much more readable.

If we pre-compose the meaning function ⟦_⟧ with a notion of 'de Bruijn scopes' (denoted Scope here) which turns any I –Scoped family into a function of type List $I \to I$ –Scoped by appending the two List indices, we recover a meaning function producing an endofunctor on I –Scoped. So far we have only shown the action of the functor on objects; its action on morphisms is given by a function fmap defined by induction over the description just as in Section 4.

```
Scope : I –Scoped → List I → I –Scoped
Scope T Δ i = (Δ ++_) ⊢ T i
```

Fig. 16. De Bruijn Scopes

The endofunctors thus defined are strictly positive and we can take their fixpoints. As we want to define the terms of a language with variables, instead of considering the initial algebra, this time we opt for the free relative monad (Altenkirch *et al.* (2014)) (with respect to the functor Var): the 'var constructor corresponds to return, and we will define bind (also known as the parallel substitution sub) in the next section.

```
data Tm (d : Desc I) : Size → I –Scoped where
  'var : ∀[ Var i                          ⇒ Tm d (↑ s) i ]
  'con : ∀[ ⟦ d ⟧ (Scope (Tm d s)) i ⇒ Tm d (↑ s) i ]
```

Fig. 17. Term Trees: The Free Var-Relative Monads on Descriptions

Coming back to our original examples, we now have the ability to give codes for the well scoped untyped λ-calculus and, just as well, the intrinsically typed simply typed λ-calculus. We add a third example to showcase the whole spectrum of syntaxes: a well scoped and well sorted but not well typed bidirectional language. In all examples, the variable case will be added by the free monad construction so we only have to describe the other constructors.

Untyped λ-calculus For the untyped case, the lack of type translates to picking unit as our notion of sort. We have two possible constructors: application where we have two substructures which do not bind any extra argument and λ-abstraction which has exactly one substructure with precisely one extra bound variable. A single Boolean is enough to distinguish the two constructors.

```
UTLC : Desc ⊤
UTLC = 'σ Bool $ λ isApp → if isApp
  then 'X [] tt ('X [] tt ('■ tt))
  else 'X (tt :: []) tt ('■ tt)
```

Fig. 18. Description for the Untyped λ-calculus

Bidirectional STLC Our second example is a bidirectional (Pierce & Turner (2000)) language hence the introduction of a notion of Mode: each term is either part of the Infer or Check fraction of the language. This language has four constructors which we list in the ad-hoc ‘Bidi type of constructor tags, its decoding Bidi is defined by a pattern-matching λ-expression in Agda. Application and λ-abstraction behave as expected, with the important observation that λ-abstraction binds an Inferrable term. The two remaining constructors correspond to changes of direction: one can freely Embbed inferrable terms as checkable ones whereas we require a type annotation when forming a Cut.

```
data Mode : Set where
  Check Infer : Mode

data ‘Bidi : Set where
  App Lam Emb : ‘Bidi
  Cut : Type → ‘Bidi

Bidi : Desc Mode
Bidi = ‘σ ‘Bidi $ λ where
  App     → ‘X [] Infer (‘X [] Check (‘■ Infer))
  Lam     → ‘X (Infer :: []) Check (‘■ Check)
  (Cut σ) → ‘X [] Check (‘■ Infer)
  Emb     → ‘X [] Infer (‘■ Check)
```

Fig. 19. Description for the bidirectional STLC

Intrinsically typed STLC In the typed case, we are back to two constructors: the terms are fully annotated and therefore it is not necessary to distinguish between Modes anymore. We need our tags to carry extra information about the types involved so we use once more and ad-hoc datatype ‘STLC, and define its decoding STLC by a pattern-matching λ-expression.

```
data ‘STLC : Set where
  App Lam : Type → Type → ‘STLC

STLC : Desc Type
STLC = ‘σ ‘STLC $ λ where
  (App σ τ) → ‘X [] (σ ‘→ τ) (‘X [] σ (‘■ τ))
  (Lam σ τ) → ‘X (σ :: []) τ (‘■ (σ ‘→ τ))
```

Fig. 20. Description for the intrinsically typed STLC

For convenience we use Agda’s pattern synonyms corresponding to the original constructors in Section 2. These synonyms can be used when pattern-matching on a term and Agda resugars them when displaying a goal. This means that the end user can seamlessly work with encoded terms without dealing with the gnarly details of the encoding. These pattern definitions can omit some arguments by using “_”, in which case they will be filled in by unification just like any other implicit argument: there is no extra cost to using an encoding! The only downside is that the language currently does not allow the user to specify type annotations for pattern synonyms. We only include examples of pattern synonyms for the two extreme examples, the definition for Bidi are similar.

```
pattern ‘app f t = ‘con (true , f , t , refl)
pattern ‘lam b   = ‘con (false , b , refl)

pattern ‘app f t = ‘con (App _ _ , f , t , refl)
pattern ‘lam b   = ‘con (Lam _ _ , b , refl)
```

Fig. 21. Respective Pattern Synonyms for UTLC and STLC.

As a usage example of these pattern synonyms, we define the identity function in all three languages in Figure 22. The code is virtually the same except for Bidi which records explicitly the change of direction from Check to Infer.

```
id^U : Tm UTLC ∞ tt []          id^B : Tm Bidi ∞ Check []          id^S : Tm STLC ∞ (σ ‘→ σ) []
id^U = ‘lam (‘var z)            id^B = ‘lam (‘emb (‘var z))        id^S = ‘lam (‘var z)
```

Fig. 22. Identity function in all three languages

It is the third time (the first and second times being the definition of listD and vecD in Figure 9 and 10) that we use a Bool to distinguish between two constructors. In order to avoid re-encoding the same logic, the next section introduces combinators demonstrating that descriptions are closed under finite sums.

Common Combinators and Their Properties. As seen previously, we can use a dependent pair whose first component is a Boolean to take the coproduct of two descriptions: depending on the value of the first component, we will return one or the other. We can abstract this common pattern as a combinator _‘+_ together with an appropriate eliminator case which, given two continuations, picks the one corresponding to the chosen branch.

```
_‘+_ : Desc I → Desc I → Desc I
d ‘+ e = ‘σ Bool $ λ isLeft →
           if isLeft then d else e

case : (⟦ d ⟧ X i Γ → A) → (⟦ e ⟧ X i Γ → A) →
       (⟦ d ‘+ e ⟧ X i Γ → A)
case l r (true  , t) = l t
case l r (false , t) = r t
```

Fig. 23. Descriptions are closed under Sum

A concrete use case for this combinator will be given in section 7.5 where we explain how to seamlessly enrich an existing syntax with let-bindings and how to use the Semantics framework to elaborate them away.

6 Generic Scope Safe and Well Kinded Programs for Syntaxes

Based on the Semantics type we defined for the specific example of the simply typed λ-calculus in Section 3, we can define a generic notion of semantics for all syntax descriptions. It is once more parametrised by two I–Scoped families $\mathcal{V}$ and $\mathcal{C}$ corresponding respectively to values associated to bound variables and computations delivered by evaluating terms. These two families have to abide by three constraints:

- th^$\mathcal{V}$ Values should be thinnable so that we can push the evaluation environment under binders;
- var Values should embed into computations for us to be able to return the value associated to a variable as the result of its evaluation;
- alg We should have an algebra turning a term whose substructures have been replaced with computations (possibly under some binders, represented semantically by the Kripke type-valued function defined below) into computations

```
record Semantics (d : Desc I) (𝒱 C : I −Scoped) : Set where
  field th^𝒱 : Thinnable (𝒱 σ)
        var   : ∀[ 𝒱 σ ⇒ C σ ]
        alg   : ∀[ ⟦ d ⟧ (Kripke 𝒱 C) σ ⇒ C σ ]
```

Fig. 24. A Generic Notion of Semantics

Here we crucially use the fact that the meaning of a description is defined in terms of a function interpreting substructures which has the type List $I \rightarrow I$–Scoped, i.e. that gets access to the current scope but also the exact list of the newly bound variables' kinds. We define a function Kripke by case analysis on the number of newly bound variables. It is essentially a subcomputation waiting for a value associated to each one of the fresh variables.

- If it's 0 we expect the substructure to be a computation corresponding to the result of the evaluation function's recursive call;
- But if there are newly bound variables then we expect to have a function space. In any context extension, it will take an environment of values for the newly-bound variables and produce a computation corresponding to the evaluation of the body of the binder.

```
Kripke : (𝒱 C : I −Scoped) → (List I → I −Scoped)
Kripke 𝒱 C [] j = C j
Kripke 𝒱 C Δ j = □ ((Δ −Env) 𝒱 ⇒ C j)
```

Fig. 25. Substructures as either Computations or Kripke Function Spaces

It is once more the case that the abstract notion of Semantics comes with a fundamental lemma: all I –Scoped families $\mathcal{V}$ and C satisfying the three criteria we have put forward give rise to an evaluation function. We introduce a notion of computation _–Comp analogous to that of environments: instead of associating values to variables, it associates computations to terms.

```
_–Comp : List I → I −Scoped → List I → Set
(Γ −Comp) C Δ = ∀ {s σ} → Tm d s σ Γ → C σ Δ
```

6.1 Fundamental Lemma of Semantics

We can now define the type of the fundamental lemma (called semantics) which takes a semantics and returns a function from environments to computations. It is defined mutually with a function body turning syntactic binders into semantics binders: to each de Bruijn Scope (i.e. a substructure in a potentially extended context) it associates a Kripke (i.e. a subcomputation expecting a value for each newly bound variable).

The proof of semantics is straightforward now that we have clearly identified the problem structure and the constraints we need to enforce. If the term considered is a variable, we

```
semantics : (Γ –Env) 𝒱 Δ → (Γ –Comp) 𝒞 Δ
body : (Γ –Env) 𝒱 Δ → ∀ Θ σ →
        Scope (Tm d s) Θ σ Γ → Kripke 𝒱 𝒞 Θ σ Δ
```

Fig. 26. Statement of the Fundamental Lemma of Semantics

lookup the associated value in the evaluation environment and turn it into a computation using var. If it is a non variable constructor then we call fmap to evaluate the substructures using body and then call the algebra to combine these results.

```
semantics ρ (‘var k) = var (lookup ρ k)
semantics ρ (‘con t) = alg (fmap d (body ρ) t)
```

Fig. 27. Proof of the Fundamental Lemma of Semantics – semantics

The auxiliary lemma body distinguishes two cases. If no new variable has been bound in the recursive substructure, it is a matter of calling semantics recursively. Otherwise we are provided with a Thinning, some additional values and evaluate the substructure in the thinned and extended evaluation environment (thanks to a auxiliary function _>>_ which given two environments (Γ –Env) 𝒱 Θ and (Δ –Env) 𝒱 Θ produces an environment ((Γ ++ Δ) –Env) 𝒱 Θ).

```
body ρ []        i t = semantics ρ t
body ρ (_ :: _) i t = λ σ vs → semantics (vs >> th^Env th^𝒱 ρ σ) t
```

Fig. 28. Proof of the Fundamental Lemma of Semantics – body

Given that fmap introduces one level of indirection between the recursive calls and the subterms they are acting upon, the fact that our terms are indexed by a Size is once more crucial in getting the termination checker to see that our proof is indeed well founded.

We immediately introduce a corollary of the fundamental lemma of semantics for the special cases of closed terms: we can use an empty environment to kickstart the evaluation.

6.2 Our First Generic Programs: Renaming and Substitution

Similarly to ACMM (2017) renaming can be defined generically for all syntax descriptions as a semantics with Var as values and Tm as computations. The first two constraints on Var described earlier are trivially satisfied. Observing that renaming strictly respects the structure of the term it goes through, it makes sense for the algebra to be implemented using fmap. When dealing with the body of a binder, we ‘reify’ the Kripke function by evaluating it in an extended context and feeding it placeholder values corresponding to the extra variables introduced by that context. This is reminiscent both of what we did in Section 3 and the definition of reification in the setting of normalisation by evaluation (see e.g. Coquand’s work (2002)).

```
closed : TM d σ → C σ []
closed = semantics ε
```

Fig. 29. Corollary: evaluation of closed terms

Substitution is defined in a similar manner with Tm as both values and computations. Of the two constraints applying to terms as values, the first one corresponds to renaming and the second one is trivial. The algebra is once more defined by using fmap and reifying the bodies of binders.

```
Ren : Semantics d Var (Tm d ∞)
Ren .th^𝒱 = th^Var
Ren .var   = ‘var
Ren .alg   = ‘con ∘ fmap d (reify vl^Var)

ren : (Γ −Env) Var Δ →
      Tm d ∞ σ Γ → Tm d ∞ σ Δ
ren ρ t = Semantics.semantics Ren ρ t
```

```
Sub : Semantics d (Tm d ∞) (Tm d ∞)
Sub .th^𝒱 = th^Tm
Sub .var   = id
Sub .alg   = ‘con ∘ fmap d (reify vl^Tm)

sub : (Γ −Env) (Tm d ∞) Δ →
      Tm d ∞ σ Γ → Tm d ∞ σ Δ
sub ρ t = Semantics.semantics Sub ρ t
```

Fig. 30. Generic Renaming and Substitution for All Scope Safe Syntaxes with Binding

The reification process mentioned in the definition of renaming and substitution can be implemented generically for Semantics families which have VarLike values (vl^Var and vl^Tm are proofs of VarLike for Var and Tm respectively) i.e. values which are thinnable and such that we can craft placeholder values in non-empty contexts.

```
record VarLike (𝒱 : I −Scoped) : Set where
  field th^𝒱 : Thinnable (𝒱 σ)
        new  : ∀[ (σ ::_) ⊢ 𝒱 σ ]
```

Fig. 31. VarLike: Thinnable and with placeholder values

For any VarLike $\mathcal{V}$, we can define fresh^r of type (Γ −Env) $\mathcal{V}$ (Δ ++ Γ) and fresh^l of type (Γ −Env) $\mathcal{V}$ (Γ ++ Δ) by combining the use of placeholder values and thinnings, and it is almost immediate that variables are VarLike. Hence, we can then implement reify as in Figure 32:

7 A Catalogue of Generic Programs for Syntax with Binding

In this section we explore a large part of the spectrum of traversals a compiler writer may need when implementing their own language. In Section 7.1 we look at the production of human-readable representations of internal syntax; in Section 7.2 we write a generic scope checker thus bridging the gap between raw data fresh out of a parser to well scoped syntax; we then demonstrate how to write a type checker in Section 7.3 and even an elaboration function turning well scoped into well scoped and typed syntax in Section 7.4. We then study

```
reify : VarLike 𝒱 → ∀ Δ i → Kripke 𝒱 C Δ i Γ → Scope C Δ i Γ
reify vl^𝒱 []           i b = b
reify vl^𝒱 Δ@(_ :: _) i b = b (freshʳ vl^Var Δ) (freshˡ vl^𝒱 _)
```

Fig. 32. Generic Reification thanks to VarLike Values

type and scope respecting transformations on internal syntax: desugaring in Section 7.5 and size preserving inlining in Section 7.6. We conclude with an unsafe but generic evaluator defined using normalisation by evaluation in Section 7.7.

7.1 Printing with Names

We have seen in Section 3.3 that printing with names is an instance of ACMM's notion of Semantics. We will now show that this observation can be generalised to arbitrary syntaxes with binding. Unlike renaming or substitution, this generic program will require user guidance: there is no way for us to guess how an encoded term should be printed. We can however take care of the name generation, deal with variable binding, and implement the traversal generically. We want our printer to have type:

```
print : Display d → Tm d i σ Γ → String
```

where Display explains how to print one 'layer' of term provided that we are handed the Pieces corresponding to the printed subterm and names for the bound variables where assigned:

```
Display : Desc I → Set
Display d = ∀ {i Γ} → ⟦ d ⟧ Pieces i Γ → String
```

Reusing the notion of Name introduced in Section 3.3, we can make Pieces formal. A subterm has already been printed if we have a string representation of it together with an environment of Names we have attached to the newly-bound variables this structure contains. That is to say:

```
Pieces : List I → I −Scoped
Pieces [] i Γ = String
Pieces Δ i Γ = (Δ −Env) Name (Δ ++ Γ) × String
```

The key observation that will help us define a generic printer is that M composed with Name is VarLike. Indeed as the composition of a functor and a trivially thinnable Wrapper it is Thinnable, and fresh (defined in Figure 6) is the proof that we can generate placeholder values thanks to the name supply.

```
vl^MName : VarLike {I} (λ σ → M ∘ (Name σ))
vl^MName = record
  { th^𝒱 = th^Functor functor^M th^Wrap
  ; new   = fresh _
  }
```

This VarLike instance empowers us to reify in an effectful manner a Kripke function space taking Names and returning a Printer to a set of Pieces.

```
reify^M : ∀ Δ i → Kripke Name Printer Δ i Γ → M (Pieces Δ i Γ)
```

In case there are no newly bound variables, the Kripke function space collapses to a mere Printer which is precisely the wrapped version of the type we expect.

```
reify^M [] i p = getW p
```

Otherwise we proceed in a manner reminiscent of the pure reification function defined in Figure 32. We start by generating an environment of names for the newly-bound variables by using the fact that M composed with Name is VarLike together with the fact that environments are traversable (McBride & Paterson (2008)). We then run the Kripke function on these names to obtain the string representation of the subterm. We finally return the names we used together with this string.

```
reify^M Δ@(_ :: _) i f = do
  ρ ← sequenceA (fresh^l vl^MName _)
  b ← getW (f (fresh^r vl^Var Δ) ρ)
  return (ρ , b)
```

We can put all of these pieces together to obtain the Printing semantics presented in Figure 33. The two first constraints can be trivially discharged. When defining the algebra we start by reifying the subterms, then use the fact that one “layer” of term of our syntaxes with binding is always traversable to combine all of these results into a value we can apply our display function to.

```
Printing : Display d → Semantics d Name Printer
Printing dis .th^𝒱 = th^Wrap
Printing dis .var   = map^Wrap return
Printing dis .alg   = λ v → MkW $ dis <$> mapA d reify^M v
```

Fig. 33. Printing with Names as a Semantics

This allows us to write a printer for open terms as demonstrated in Figure 34. We start by generating an environment of Names for the free variables, then use our semantics to get a printer which we can run using a stream names of distinct strings as our name supply.

```
print : Display d → Tm d i σ Γ → String
print dis t = proj₁ (printer names) where
  printer : M String
  printer = do
    init ← sequenceA (base vl^MName)
    getW (Semantics.semantics (Printing dis) init t)
```

Fig. 34. Generic Printer for Open Terms

Untyped λ-calculus Defining a printer for the untyped λ-calculus is now very easy: we define a Display by case analysis. In the application case, we combine the string representation of the function, wrap its argument's representation between parentheses and concatenate the two together. In the lambda abstraction case, we are handed the name the bound variable was assigned together with the body's representation; it is once more a matter of putting the Pieces together.

```
printUTLC : Display UTLC
printUTLC = λ where
  (‘app’ f t)      → f ++ " (" ++ t ++ ")"
  (‘lam’ (x , b)) → "λ" ++ getW (lookup x z) ++ ". " ++ b
```

As always, these functions are readily executable and we can check their behaviour by writing tests. First, we print the identity function defined in Figure 22 in an empty context and verify that we do obtain the string "λa. a". Next, we print an open term in a context of size two and can immediately observe that names are generated for the free variables first, and then the expression itself is printed.

```
_ : print printUTLC id^U ≡ "λa. a"
_ = refl
```

```
_ : let tm : Tm UTLC _ _ (_ :: _ :: [])
        tm = ‘app (‘var z) (‘lam (‘var (s (s z))))
    in print printUTLC tm ≡ "b (λc. a)"
_ = refl
```

7.2 Writing a Generic Scope Checker

Converting terms in the internal syntax to strings which can in turn be displayed in a terminal or an editor window is only part of a compiler's interaction loop. The other direction takes strings as inputs and attempts to produce terms in the internal syntax. The first step is to parse the input strings into structured data, the second is to perform scope checking, and the third step consist in type checking.

Parsing is currently out of scope for our library; users can write safe ad-hoc parsers for their object language by either using a library of total parser combinators (Danielsson (2010); Allais (2018)) or invoking a parser generator oracle whose target is a total language (Stump (2016)). As we will see shortly, we can write a generic scope checker transforming terms in a raw syntax where variables are represented as strings into a well scoped syntax. We will come back to typechecking with a concrete example in section 7.3 and then discuss related future work in the conclusion.

Our scope checker will be a function taking two explicit arguments: a name for each variable in scope Γ, a raw term for a syntax description d and either fail (the Monad M granting us the ability to fail is made explicit in Figure 36) or return a well scoped and sorted term for that description.

```
toTm : Names Γ → Raw d i σ → M (Tm d i σ Γ)
```

Raw Terms We can obtain Names by reusing the standard library's All, a predicate transformer making sure a predicate holds of all the element of a list. The definition of WithNames is analogous to Pieces in the previous section: we expect Names for the newly bound variables. Terms in the raw syntax then leverage these definitions: they are either a variable i.e. a String potentially accompanied by some extra information E (typically a position in a file) or a "layer" of raw term where subterms are raw terms with names for the newly-bound variables.

```
Names : List I → Set
Names = All (const String)

WithNames : (I → Set) → List I → I ─Scoped
WithNames T [] j Γ = T j
WithNames T Δ j Γ = Names Δ × T j

data Raw (d : Desc I) : Size → I → Set where
  'var  : E → String → Raw d (↑ i) σ
  'con : ⟦ d ⟧ (WithNames (Raw d i)) σ [] → Raw d (↑ i) σ
```

Fig. 35. Names and Raw Terms

Error Handling Various things can go wrong during scope checking: evidently a name can be out of scope but it is also possible that it may be associated to a variable of the wrong sort. We define an enumerating type covering these two cases. The scope checker will return a computation in the Monad M thus allowing us to fail and return an error, the string that caused the failure and the extra data of type E that accompanied it.

```
data Error : Set where
  OutOfScope : Error
  WrongSort   : (σ τ : I) → σ ≢ τ → Error

M : Set → Set
M A = (Error × E × String) ⊎ A
```

Fig. 36. Error Type and Scope Checking Monad

Equipped with these notions, we can write down the type of toVar which tackles the core of the problem: variable resolution. Given a string corresponding to a variable's name and provided that we have names for the variables in the ambient scope and know the sort the variable needs to be we can either fail or return a well scoped and well sorted Var.

If the ambient scope is empty then we can only fail with an OutOfScope error. Alternatively, if the variable's name corresponds to that of the first one in scope we check that the sorts match up and either return z or fail with a WrongSort error. Otherwise we look for the variable further down the scope and use s to lift the result to the full scope.

```
toVar : E → String → ∀ σ Γ → Names Γ → M (Var σ Γ)
toVar e x σ [] [] = inj₁ (OutOfScope , e , x)
toVar e x σ (τ :: Γ) (y :: scp) with x ≟ y | I-dec σ τ
... | yes _  | yes refl = inj₂ z
... | yes _  | no ¬eq  = inj₁ (WrongSort σ τ ¬eq , e , x)
... | no ¬p | _          = s <$> toVar e x σ Γ scp
```

Fig. 37. Variable Resolution

Scope checking an entire term then amounts to lifting this action on variables to an action on terms. The error Monad M is by definition an Applicative and by design our terms are Traversable (Bird & Paterson (1999); Gibbons & d. S. Oliveira (2009)). The action on term is defined mutually with the action on scopes. As we can see in the second equation for toScope, thanks to the definition of WithNames, concrete names arrive just in time to check the subterm with newly-bound variables.

```
toTm    : Names Γ → Raw d i σ → M (Tm d i σ Γ)
toScope : Names Γ → ∀ Δ σ → WithNames (Raw d i) Δ σ [] → M (Scope (Tm d i) Δ σ Γ)

toTm scp (‘var e v) = ‘var <$> toVar e v _ _ scp
toTm scp (‘con b)   = ‘con <$> mapA d (toScope scp) b

toScope scp []            σ b        = toTm scp b
toScope scp Δ@(_ :: _) σ (bnd , b) = toTm (bnd ++ scp) b
```

Fig. 38. Generic Scope Checking for Terms and Scopes

7.3 An Algebraic Approach to Typechecking

Following Atkey (2015), we can consider type checking and type inference as a possible semantics for a bi-directional (Pierce & Turner (2000)) language. We reuse the syntax introduced in Section 5; it gives us a simply typed bi-directional calculus as a bi-sorted language using a notion of Mode to distinguish between terms for which we will be able to Infer the type and the ones for which we will have to Check a type candidate.

The values stored in the environment of the typechecking function attach Type information to bound variables whose Mode is Infer, guaranteeing no variable ever uses the Check mode. In contrast, the generated computations will, depending on the mode, either take a type candidate and Check it is valid or Infer a type for their argument. These computations are always potentially failing so we use the Maybe monad.

```
data Var- : Mode → Set where
  ‘var : Type → Var- Infer

Type- : Mode → Set
Type- Check = Type → Maybe ⊤
Type- Infer =        Maybe Type
```

Fig. 39. Var- and Type- Relations indexed by the Mode

A change of direction from inferring to checking will require being able to check that two types agree so we introduce the function _=?_. Similarly we will sometimes expect a function type but may be handed anything so we will have to check with isArrow that our candidate's head constructor is indeed an arrow, and collect the domain and codomain.

We can now define typechecking as a Semantics. We describe the algorithm constructor by constructor; in the Semantics definition (omitted here) the algebra will simply perform a dispatch and pick the relevant auxiliary lemma. Note that in the following code, _<$_ is, following classic Haskell notations, the function which takes an A and a Maybe B and returns a Maybe A which has the same structure as its second argument.

```
_=?_ : (σ τ : Type) → Maybe ⊤
α          =? α          = just tt
(σ ‘→ τ) =? (ϕ ‘→ ψ) = (σ =? ϕ) >> (τ =? ψ)
_          =? _          = nothing
```

```
isArrow : Type → Maybe (Type × Type)
isArrow (σ ‘→ τ) = just (σ , τ)
isArrow _           = nothing
```

Fig. 40. Tests for Type values

Application When facing an application: infer the type of the function, make sure it is an arrow type, check the argument at the domain's type and return the codomain.

```
app : Type- Infer → Type- Check → Type- Infer
app f t = do
  arr      ← f
  (σ , τ) ← isArrow arr
  τ <$ t σ
```

λ-abstraction For a λ-abstraction: check the input type is an arrow type and check the body at the codomain type in the extended environment where the newly-bound variable is of mode Infer and has the domain's type.

```
lam : Kripke (const ∘ Var-) (const ∘ Type-) (Infer :: []) Check Γ → Type- Check
lam b arr = do
  (σ , τ) ← isArrow arr
  b (bind Infer) (ε • ‘var σ) τ
```

Cut: A Check in an Infer position A cut always comes with a type candidate against which to check the term and to be returned in case of success.

```
cut : Type → Type- Check → Type- Infer
cut σ t = σ <$ t σ
```

Embedding of Infer into Check Finally, the change of direction from Inferrable to Checkable is successful when the inferred type is equal to the expected one.

```
emb : Type- Infer → Type- Check
emb t σ = do
  τ ← t
  σ =? τ
```

We have defined a bidirectional typechecker for this simple language by leveraging the Semantics framework. We can readily run it on closed terms using the closed corollary defined in Figure 29 and (defining β to be (α ‘→ α)) infer the type of the expression (λx. x : $\beta \to \beta$) (λx. x).

The output of this function is not very informative. As we will see shortly, there is nothing stopping us from moving away from a simple computation returning a (Maybe Type) to an evidence-producing function elaborating a term in Bidi to a well scoped and typed term in STLC.

ZU064-05-FPR jfp19 31 May 2019 18:22


```
type- : ∀ p → TM Bidi p → Type- p          _ : type- Infer (‘app (‘cut (β ‘→ β) id^B) id^B) ≡ just β
type- p = Semantics.closed Typecheck       _ = refl
```

Fig. 41. Type- Inference / Checking as a Semantics

7.4 An Algebraic Approach to Elaboration

Instead of generating a type or checking that a candidate will do, we can use our language of Descriptions to define not only an untyped source language but also an intrinsically typed internal language. During typechecking we generate at the same time an expression's type and a well scoped and well typed term of that type. We use STLC (defined in Section 5) as our internal language.

Before we can jump right in, we need to set the stage: a Semantics for a Bidi term will involve (Mode −Scoped) notions of values and computations but an STLC term is (Type −Scoped). We first introduce a Typing associating types to each of the modes in scope, together with fromTyping extracting the context thus defined.

```
Typing : List Mode → Set          fromTyping : Typing ms → List Type
Typing = All (const Type)         fromTyping []        = []
                                  fromTyping (σ :: Γ) = σ :: fromTyping Γ
```

Fig. 42. Typing: From Contexts of Modes to Contexts of Types

We can then explain what it means for elaboration to target T a (Type −Scoped) at a type σ: provided a list of modes and a corresponding Typing, we should get a T of type σ in the typing context induced by that Typing.

```
Elab : Type −Scoped → Type → (ms : List Mode) → Typing ms → Set
Elab T σ _ Γ = T σ (fromTyping Γ)
```

Fig. 43. Elaboration of a Scoped Family

Our first example of an elaboration function is our notion of environment values. To each variable in scope of mode Infer we associate an elaboration function targeting Var. In other words: our values are all in scope i.e. provided any typing of the scope, we can assuredly return a type together with a variable of that type.

```
data Var- : Mode −Scoped where
  ‘var : (infer : ∀ Γ → Σ[ σ ∈ Type ] Elab Var σ ms Γ) → Var- Infer ms
```

Fig. 44. Values as Elaboration Functions for Variables

We can for instance prove that we have such an inference function for a newly-bound variable of mode Infer: given that the context has been extended with a variable of mode Infer, the Typing must also have been extended with a type σ. We can return that type paired with the variable z.

```
var₀ : Var- Infer (Infer :: ms)
var₀ = ‘var λ where (σ :: _) → (σ , z)
```

Fig. 45. Inference Function for the 0-th Variable

The computations are a bit more tricky. On the one hand, if we are in checking mode then we expect that for any typing of the scope of modes and any type candidate we can Maybe return a term at that type in the induced context. On the other hand, in the inference mode we expect that given any typing of the scope, we can Maybe return a type together with a term at that type in the induced context.

```
Elab- : Mode −Scoped
Elab- Check ms = ∀ Γ → (σ : Type) → Maybe (Elab (Tm STLC ∞) σ ms Γ)
Elab- Infer   ms = ∀ Γ → Maybe (Σ[ σ ∈ Type ] Elab (Tm STLC ∞) σ ms Γ)
```

Fig. 46. Computations as Mode-indexed Elaboration Functions

Because we are now writing a typechecker which returns evidence of its claims, we need more informative variants of the equality and isArrow checks. In the equality checking case we want to get a proof of propositional equality but we only care about the successful path and will happily return nothing when failing. Agda's support for (dependent!) do-notations makes writing the check really easy. For the arrow type, we introduce a family Arrow constraining the shape of its index to be an arrow type and redefine isArrow as a *view* targeting this inductive family (Wadler (1987); McBride & McKinna (2004)).

```
_=?_ : (σ τ : Type) → Maybe (σ ≡ τ)
α           =? α           = just refl
(σ ‘→ τ) =? (ϕ ‘→ ψ) = do
  refl ← σ =? ϕ
  refl ← τ =? ψ
  return refl
_ =? _ = nothing
```

```
data Arrow : Type → Set where
  _‘→_ : ∀ σ τ → Arrow (σ ‘→ τ)

isArrow : ∀ σ → Maybe (Arrow σ)
isArrow (σ ‘→ τ) = just (σ ‘→ τ)
isArrow _           = nothing
```

Fig. 47. Informative Equality Check and Arrow View

We now have all the basic pieces and can start writing elaboration code. We once more start by dealing with each constructor in isolation before putting everything together to get a Semantics. These steps are very similar to the ones in the previous section.

Application In the application case, we start by elaborating the function and we get its type together with its internal representation. We then check that the inferred type is indeed an Arrow and elaborate the argument using the corresponding domain. We conclude by returning the codomain together with the internal function applied to the internal argument.

```
app : ∀[ Elab- Infer ⇒ Elab- Check ⇒ Elab- Infer ]
app f t Γ = do
```

```
  (arr , F) ← f Γ
  (σ ‘→ τ) ← isArrow arr
  T         ← t Γ σ
  return (τ , ‘app F T)
```

λ-abstraction For the λ-abstraction case, we start by checking that the type candidate is an Arrow. We can then elaborate the body of the lambda in a context extended with one Infer variable assigned an inference function thanks to the auxiliary function var_0. From this we get an internal term corresponding to the body of the λ-abstraction and conclude by returning it wrapped in a ‘lam constructor.

```
lam : ∀[ Kripke Var- Elab- (Infer :: []) Check ⇒ Elab- Check ]
lam b Γ arr = do
  (σ ‘→ τ) ← isArrow arr
  B         ← b (bind Infer) (ε • var₀) (σ :: Γ) τ
  return (‘lam B)
```

Cut: A Check in an Infer position For cut, we start by elaborating the term with the type annotation provided and return them paired together.

```
cut : Type → ∀[ Elab- Check ⇒ Elab- Infer ]
cut σ t Γ = (σ ,_) <$> t Γ σ
```

Embedding of Infer into Check For the change of direction Emb we not only want to check that the inferred type and the type candidate are equal: we need to cast the internal term labelled with the inferred type to match the type candidate. Luckily, Agda's dependent do-notations make once again our job easy: when we make the pattern refl explicit, the equality holds in the rest of the block.

```
emb : ∀[ Elab- Infer ⇒ Elab- Check ]
emb t Γ σ = do
  (τ , T) ← t Γ
  refl     ← σ =? τ
  return T
```

We have almost everything we need to define elaboration as a semantics. Discharging the th^𝒱 constraint is a bit laborious and the proof doesn't yield any additional insight so we leave it out here. The semantical counterpart of variables (var) is fairly straightforward: provided a Typing, we run the inference and touch it up to return a term rather than a mere variable. Finally we define the algebra (alg) by pattern-matching on the constructor and using our previous combinators.

We can once more define a specialised version of the traversal induced by this Semantics for closed terms: not only can we give a (trivial) initial environment (using the closed corollary defined in Figure 29) but we can also give a (trivial) initial Typing. This leads to the following definitions:

```
Type- : Mode → Set                                 type- : ∀ p → TM Bidi p → Type- p
Type- Check = ∀ σ → Maybe (TM STLC σ)              type- Check t = closed Elaborate t []
Type- Infer   = Maybe (∃ λ σ → TM STLC σ)          type- Infer    t = closed Elaborate t []
```

Fig. 48. Evidence-producing Type (Checking / Inference) Function

Revisiting the example introduced in Section 7.3, we can check that elaborating the expression (λx. x : β → β) (λx. x) yields the type β together with the term (λx. x) (λx. x) in internal syntax. Type annotations have disappeared in the internal syntax as all the type invariants are enforced intrinsically.

```
_ :  type- Infer ( B.‘app (B.‘cut (β ‘→ β) id^B) id^B)
   ≡ just (β       , S.‘app                  id^S  id^S)
_ = refl
```

7.5 Sugar and Desugaring as a Semantics

One of the advantages of having a universe of programming language descriptions is the ability to concisely define an *extension* of an existing language by using Description transformers grafting extra constructors à la Swiestra (2008). This is made extremely simple by the disjoint sum combinator _‘+_ which we defined in Section 5. An example of such an extension is the addition of let-bindings to an existing language.

Let bindings allow the user to avoid repeating themselves by naming sub-expressions and then using these names to refer to the associated terms. Preprocessors adding these types of mechanisms to existing languages (from C to CSS) are rather popular. We introduce a description of Let-bindings which can be used to extend any language description *d* to *d* ‘+ Let (where ‘+ is the disjoint sum of two descriptions defined in Figure 23):

```
Let : Desc I                                  pattern ‘IN’ e t = (_ , e , t , refl)
Let = ‘σ (I × I) $ uncurry $ λ σ τ →          pattern ‘IN e t = ‘con (‘IN’ e t)
       ‘X [] σ (‘X (σ :: []) τ (‘■ τ))
```

Fig. 49. Description of a single let binding, associated pattern synonyms

This description states that a let-binding node stores a pair of types σ and τ and two subterms. First comes the let-bound expression of type σ and second comes the body of the let which has type τ in a context extended with a fresh variable of type σ. This defines a term of type τ.

In a dependently typed language, a type may depend on a value which in the presence of let bindings may be a variable standing for an expression. The user naturally does not want it to make any difference whether they used a variable referring to a let-bound expression or the expression itself. Various typechecking strategies can accommodate this expectation: in Coq (The Coq Development Team (2017)) let bindings are primitive constructs of the language and have their own typing and reduction rules whereas in Agda they are elaborated away to the core language by inlining.

This latter approach to extending a language d with let bindings by inlining them before typechecking can be implemented generically as a semantics over (d '+ Let). For this semantics values in the environment and computations are both let-free terms. The algebra of the semantics can be defined by parts thanks to case defined in Section 5: the old constructors are kept the same by interpreting them using the generic Substitution algebra; whilst the let-binder precisely provides the extra value to be added to the environment.

```
UnLet : Semantics (d '+ Let) (Tm d ∞) (Tm d ∞)
Semantics.th^𝒱 UnLet = th^Tm
Semantics.var  UnLet = id
Semantics.alg  UnLet = case (Semantics.alg Sub) $ λ where
  ('IN' e t) → extract t (ε ∙ e)
```

Fig. 50. Desugaring as a Semantics

The process of removing let binders is kickstarted with a placeholder environment associating each variable to itself.

```
unlet : ∀[ Tm (d '+ Let) ∞ σ ⇒ Tm d ∞ σ ]
unlet = Semantics.semantics UnLet (pack 'var)
```

Fig. 51. Specialising semantics with an environment of placeholder values

In less than 10 lines of code we have defined a generic extension of syntaxes with binding together with a semantics which corresponds to an elaborator translating away this new construct. In their own setting working on STLC, ACMM (2017) have shown that it is similarly possible to implement a Continuation Passing Style transformation as a semantics.

We have demonstrated how easily one can define extensions and combine them on top of a base language without having to reimplement common traversals for each one of the intermediate representations. Moreover, it is possible to define *generic* transformations elaborating these added features in terms of lower-level ones. This suggests that this setup could be a good candidate to implement generic compilation passes and could deal with a framework using a wealth of slightly different intermediate languages à la Nanopass (Keep & Dybvig (2013)).

7.6 Reference Counting and Inlining as a Semantics

Although useful in its own right, desugaring all let bindings can lead to an exponential blow-up in code size. Compiler passes typically try to maintain sharing by only inlining let-bound expressions which appear at most one time. Unused expressions are eliminated as dead code whilst expressions used exactly one time can be inlined: this transformation is size preserving and opens up opportunities for additional optimisations.

As we will see shortly, we can implement reference counting and size respecting let-inlining as a generic transformation over all syntaxes with binding equipped with let binders. This two-pass simple transformation takes linear time which may seem surprising given the results due to Appel and Jim (1997). Our optimisation only inlines let-bound variables whereas theirs also encompasses the reduction of static β-redices of (potentially) recursive function. While we can easily count how often a variable is used in the body of a let binder, the interaction between inlining and β-reduction in theirs creates cascading simplification opportunities thus making the problem much harder.

But first, we need to look at an example demonstrating that this is a slightly subtle matter. Assuming that *expensive* takes a long time to evaluate, inlining all of the lets in the first expression is a really good idea whilst we only want to inline the one binding *y* in the second one to avoid duplicating work. That is to say that the contribution of the expression bound to *y* in the overall count depends directly on whether *y* itself appears free in the body of the let which binds it.

```
_ = let x = expensive in          _ = let x = expensive in
    let y = (x , x)   in              let y = (x , x)   in
    x                                 y
```

Our transformation will consist of two passes: the first one will annotate the tree with accurate count information precisely recording whether let-bound variables are used zero, one, or many times. The second one will inline precisely the let-binders whose variable is used at most once.

During the counting phase we need to be particularly careful not to overestimate the contribution of a let-bound expression. If the let-bound variable is not used then we can naturally safely ignore the associated count. But if it used many times then we know we will not inline this let-binding and the count should therefore only contribute once to the running total. We define the control combinator in Figure 56 precisely to explicitly handle this subtle case.

The first step is to introduce the Counter additive monoid. Addition will allow us to combine counts coming from different subterms: if any of the two counters is zero then we return the other, otherwise we know we have many occurences.

```
data Counter : Set where          _+_ : Counter → Counter → Counter
  zero  : Counter                 zero + n    = n
  one   : Counter                 m    + zero = m
  many  : Counter                 _    + _    = many
```

Fig. 52. The (Counter, zero, _+_) additive monoid

The annotation process can be described as a variation on the Let syntax extension of Section 7.5, attaching a Counter to each Let node. That is to say, a function computing a (*d* ‘+ CLet) term from a (*d* ‘+ Let) one.

We keep a tally of the usage information for the variables in scope. This allows us to know which Counter to attach to each Let node. Following the same strategy as in Section 7.2, we use the standard library’s All to represent this mapping. We say that a scoped value has been Counted if it is paired with a Count.

```
CLet : Desc I
CLet = ‘σ Counter $ λ _ → Let
```

Fig. 53. Counted Lets

```
Count : List I → Set                    Counted : I −Scoped → I −Scoped
Count = All (const Counter)             Counted T i Γ = T i Γ × Count Γ
```

Fig. 54. Counting i.e. Associating a Counter to each Var in scope.

The two most basic counts are described in Figure 55: the empty one is zero everywhere and the one corresponding to a single use of a single variable v which is zero everywhere except for v where it's one.

```
zeros : ∀[ Count ]                      fromVar : ∀[ Var σ ⇒ Count ]
zeros {[]}     = []                     fromVar z     = one :: zeros
zeros {σ :: Γ} = zero :: zeros          fromVar (s v) = zero :: fromVar v
```

Fig. 55. Zero Count and Count of One for a Specific Variable

When we collect usage information from different subterms, we need to put the various counts together. The combinators in Figure 56 allow us to easily do so: merge adds up two counts in a pointwise manner while control uses one Counter to decide whether to erase an existing Count. This is particularly convenient when computing the contribution of a let-bound expression to the total tally: the contribution of the let-bound expression will only matter if the corresponding variable is actually used.

```
merge : ∀[ Count ⇒ Count ⇒ Count ]      control : Counter → ∀[ Count ⇒ Count ]
merge []       []       = []            control zero cs = zeros
merge (m :: cs) (n :: ds) =             control one  cs = cs -- inlined
  (m + n) :: merge cs ds                control many cs = cs -- not inlined
```

Fig. 56. Combinators to Compute Counts

We can now focus on the core of the elaboration phase. We define a Semantics whose values are variables themselves and computations are the pairing of a term in (d ‘+ CLet) together with a Count. The variable case is trivial: provided a variable v, we return (‘var v) together with the count (fromVar v).

The non-let case is purely structural: we reify the Kripke function space and obtain a scope together with the corresponding Count. We unceremoniously drop the Counters associated to the variables bound in this subterm and return the scope together with the tally for the ambient context.

The Let-to-CLet case in Figure 58 is the most interesting one. We start by reifying the *body* of the let binder which gives us a tally cx for the bound variable and ct for the body's contribution to the ambient environment's Count. We annotate the node with cx and use

```
reify^Count : ∀ Δ σ → Kripke Var (Counted (Tm (d '+ CLet) ∞)) Δ σ Γ →
                      Counted (Scope (Tm (d '+ CLet) ∞) Δ) σ Γ
reify^Count Δ σ kr = let (scp , c) = reify vl^Var Δ σ kr in scp , drop Δ c
```

Fig. 57. Purely Structural Case

it as a control to decide whether we are going to merge any of the let-bound's expression contribution *ce* to form the overall tally.

```
clet : ⟦ Let ⟧ (Kripke Var (Counted (Tm (d '+ CLet) ∞))) σ Γ →
       Counted (⟦ CLet ⟧ (Scope (Tm (d '+ CLet) ∞))) σ Γ
clet (στ , (e , ce) , body , eq) = case body extend (ε • z) of λ where
  (t , cx :: ct) → (cx , στ , e , t , eq) , merge (control cx ce) ct
```

Fig. 58. Annotating Let Binders

Putting all of these things together we obtain the Semantics Annotate which we promptly specialise using an environment of placeholder value to obtain the traversal annotate elaborating raw let-binders into counted ones.

```
annotate : Tm (d '+ Let) ∞ σ Γ → Tm (d '+ CLet) ∞ σ Γ
annotate = proj₁ ∘ Semantics.semantics Annotate (base vl^Var)
```

Fig. 59. Specialising semantics to obtain an annotation function

Using techniques similar to the ones described in Section 7.5, we can write an Inline semantics working on (*d* '+ CLet) terms and producing (*d* '+ Let) ones. We make sure to preserve all the let-binders annotated with many and to inline all the other ones. By composing Annotate with Inline we obtain a size-preserving generic optimisation pass.

7.7 (Unsafe) Normalisation by Evaluation

A key type of traversal we have not studied yet is a language's evaluator. Our universe of syntaxes with binding does not impose any typing discipline on the user-defined languages and as such cannot guarantee their totality. This is embodied by one of our running examples: the untyped λ-calculus. As a consequence there is no hope for a safe generic framework to define normalisation functions.

The clear connection between the Kripke functional space characteristic of our semantics and the one that shows up in normalisation by evaluation suggests we ought to manage to give an unsafe generic framework for normalisation by evaluation. By temporarily **disabling Agda's positivity checker**, we can define a generic reflexive domain Dm (cf. Figure 60) in which to interpret our syntaxes. It has three constructors corresponding respectively to a free variable, a constructor's counterpart where scopes have become Kripke functional spaces on Dm and an error token because the evaluation of untyped programs may go wrong.

```
{-# NO_POSITIVITY_CHECK #-}
data Dm (d : Desc I) : Size → I −Scoped where
  V : ∀[ Var σ ⇒ Dm d s σ ]
  C : ∀[ ⟦ d ⟧ (Kripke (Dm d s) (Dm d s)) σ ⇒ Dm d (↑ s) σ ]
  ⊥ : ∀[ Dm d (↑ s) σ ]
```

Fig. 60. Generic Reflexive Domain

This datatype definition is utterly unsafe. The more conservative user will happily restrict themselves to typed settings where the domain can be defined as a logical predicate or opt instead for a step-indexed approach.

But this domain does make it possible to define a generic nbe semantics which, given a term, produces a value in the reflexive domain. Thanks to the fact we have picked a universe of finitary syntaxes, we can *traverse* (McBride & Paterson (2008); Gibbons & d. S. Oliveira (2009)) the functor to define a (potentially failing) reification function turning elements of the reflexive domain into terms. By composing them, we obtain the normalisation function which gives its name to normalisation by evaluation.

The user still has to explicitly pass an interpretation of the various constructors because there is no way for us to know what the binders are supposed to represent: they may stand for λ-abstractions, Σ-types, fixpoints, or anything else.

```
reify^Dm : ∀[ Dm d s σ ⇒ Maybe ∘ Tm d ∞ σ ]
nbe      : Alg d (Dm d ∞) (Dm d ∞) → Semantics d (Dm d ∞) (Dm d ∞)

norm     : Alg d (Dm d ∞) (Dm d ∞) → ∀[ Tm d ∞ σ ⇒ Maybe ∘ Tm d ∞ σ ]
norm alg = reify^Dm ∘ Semantics.semantics (nbe alg) (base vl^Dm)
```

Fig. 61. Generic Normalisation by Evaluation Framework

Using this setup, we can write a normaliser for the untyped λ-calculus by simply providing an algebra. We start by defining two patterns for semantical values: one for application and the other for lambda abstraction.

```
pattern LAM f = C (false , f , refl)
pattern APP' f t = (true , f , t , refl)
```

Fig. 62. Pattern synonyms for UTLC-specific Dm values

We then define the algebra by case analysis: if the node at hand is an application and its first component evaluates to a lambda, we fire it. Otherwise we have either a stuck application or a lambda, in other words we already have a value and can simply return it using C.

We have not used the ⊥ constructor so *if* the evaluation terminates (by disabling totality checking we have lost all guarantees of the sort) we know we will get a term in normal form.

```
norm^LC : ∀[ Tm UTLC ∞ tt ⇒ Maybe ∘ Tm UTLC ∞ tt ]
norm^LC = norm $ λ where
  (APP' (LAM f) t) → extract f (ε • t) - redex
  t                → C t               - value
```

Fig. 63. Normalisation by Evaluation for the Untyped λ-Calculus

See for instance this untyped yet normalising term: (λx. x) ((λx. x) (λx. x)) normalises to (λx. x).

```
_ : norm^LC (‘app id^U (‘app id^U id^U)) ≡ just id^U
_ = refl
```

Fig. 64. Example of a normalising untyped term

8 Other Opportunities for Generic Programming

Some generic programs of interest do not fit in the Semantics framework. They can still be implemented once and for all, and even benefit from the Semantics-based definitions.

We will first explore existing work on representing cyclic structures using a syntax with binding: a binder is a tree node declaring a pointer giving subtrees the ability to point back to it, thus forming a cycle. Substitution will naturally play a central role in giving these finite terms a semantics as their potentially infinite unfolding.

We will then see that much of the standard traversals produced by the ‘deriving’ machinery familiar to Haskell programmers can be implemented on syntaxes too, sometimes with more informative types.

8.1 Binding as Self-Reference: Representing Cyclic Structures

Ghani, Hamana, Uustalu and Vene (2006) have demonstrated how Altenkirch and Reus’ type-level de Bruijn indices (1999) can be used to represent potentially cyclic structures by a finite object. In their representation each bound variable is a pointer to the node that introduced it. Given that we are, at the top-level, only interested in structures with no “dangling pointers”, we introduce the notation TM *d* to mean closed terms (i.e. terms of type Tm *d* ∞ []).

A basic example of such a structure is a potentially cyclic list which offers a choice of two constructors: [] which ends the list and _::_ which combines a head and a tail but also acts as a binder for a self-reference; these pointers can be used by using the var constructor which we have renamed ↶ (pronounced “backpointer”) to match the domain-specific meaning. We can see this approach in action in the examples [0, 1] and 01↺ (pronounced “0-1-cycle”) which describe respectively a finite list containing 0 followed by 1 and a cyclic list starting with 0, then 1, and then repeating the whole list again by referring to the first cons cell represented here by the de Bruijn variable 1 (i.e. s z).

```
CListD : Set → Desc ⊤
CListD A = ‘■ tt
          ‘+ ‘σ A (λ _ → ‘X (tt :: []) tt (‘■ tt))

pattern []        = ‘con (true , refl)
pattern _::_ x xs = ‘con (false , x , xs , refl)
pattern ↶_ k      = ‘var k
```

```
[0,1] : TM (CListD ℕ) tt
01↻ : TM (CListD ℕ) tt

[0,1] = 0 :: 1 :: []
01↻ = 0 :: 1 :: ↶ s z
```

Fig. 65. Potentially Cyclic Lists: Description, Pattern Synonyms and Examples

These finite representations are interesting in their own right and we can use the generic semantics framework defined earlier to manipulate them. A basic building block is the unroll function which takes a closed tree, exposes its top node and unrolls any cycle which has it as its starting point. We can decompose it using the plug function which, given a closed and an open term, closes the latter by plugging the former at each free ‘var leaf. Noticing that plug's fundamental nature is that of substituting a term for each leaf, it makes sense to implement it by re-using the Substitution semantics we already have.

```
plug : TM d tt → ∀ Δ i → Scope (Tm d ∞) Δ i [] → TM d i
plug t Δ i = Semantics.semantics Sub (pack (λ _ → t))

unroll : TM d tt → ⟦ d ⟧ (Const (TM d)) tt []
unroll t@(‘con b) = fmap d (plug t) b
```

Fig. 66. Plug and Unroll: Exposing a Cyclic Tree's Top Layer

However, one thing still out of our reach with our current tools is the underlying co-finite trees these finite objects are meant to represent. We start by defining the coinductive type corresponding to them as the greatest fixpoint of a notion of layer. One layer of a co-finite tree is precisely given by the meaning of its description where we completely ignore the binding structure. We show with 01··· the infinite list that corresponds to the example 01↻ given above. The definition proceeds by copattern-matching as introduced in (Abel *et al.* (2013)) and showcased in (Thibodeau *et al.* (2016)).

```
record ∞Tm (d : Desc I) (s : Size) (i : I) : Set where
  coinductive; constructor ‘con
  field force : {s’ : Size< s} →
                ⟦ d ⟧ (Const (∞Tm d s’)) i []
```

```
01··· : ∀ {s} → ∞Tm (CListD ℕ) s tt
10··· : ∀ {s} → ∞Tm (CListD ℕ) s tt
01··· .force = false , 0 , 10··· , refl
10··· .force = false , 1 , 01··· , refl
```

Fig. 67. Co-finite Trees: Definition and Example

We can then make the connection between potentially cyclic structures and the co-finite trees formal by giving an unfold function which, given a closed term, produces its unfolding. The definition proceeds by unrolling the term's top layer and co-recursively unfolding all the subterms.

Even if the powerful notion of semantics described in Section 6 cannot encompass all the traversals we may be interested in, it provides us with reusable building blocks: the

```
unfold : TM d tt → ∞Tm d s tt
unfold t .force = fmap d (λ _ _ → unfold) (unroll t)
```

Fig. 68. Generic Unfold of Potentially Cyclic Structures

definition of unfold was made very simple by reusing the generic program fmap and the Substitution semantics whilst the definition of ∞Tm was made easy by reusing ⟦_⟧.

8.2 Generic Decidable Equality for Terms

Haskell programmers are used to receiving help from the 'deriving' mechanism (Hinze & Peyton Jones (2000); Magalhães *et al.* (2010)) to automatically generate common traversals for every inductive type they define. Recalling that generic programming is normal programming over a universe in a dependently typed language (Altenkirch & McBride (2002)), we ought to be able to deliver similar functionalities.

We will focus in this section on the definition of an equality test. The techniques used in this concrete example are general enough that they also apply to the definition of an ordering test, a Show instance, etc. In type theory we can do better than an uninformative boolean function claiming that two terms are equal: we can implement a decision procedure for propositional equality (Löh & Magalhães (2011)) which either returns a proof that its two inputs are equal or a proof that they cannot possibly be.

The notion of decidability can be neatly formalised by an inductive family with two constructors: a Set P is decidable if we can either say yes and return a proof of P or no and provide a proof of the negation of P (here, a proof that P implies the empty type ⊥).

```
data ⊥ : Set where

data Dec (P : Set) : Set where
  yes : P           → Dec P
  no  : (P → ⊥) → Dec P
```

Fig. 69. Empty Type and Decidability as an Inductive Family

To get acquainted with these new notions we can start by proving that equality of variables is decidable.

8.2.1 Deciding Variable Equality

The type of the decision procedure for equality of variables is as follows: given any two variables (of the same type, in the same context), the set of equality proofs between them is Decidable.

```
eq^Var : (v w : Var σ Γ) → Dec (v ≡ w)
```

We can easily dismiss two trivial cases: if the two variables have distinct head constructors then they cannot possibly be equal. Agda allows us to dismiss the impossible premisse of the function stored in the no contructor by using an absurd pattern ().

```
eq^Var z      (s w) = no (λ ())
eq^Var (s v) z      = no (λ ())
```

Otherwise if the two head constructors agree we can be in one of two situations. If they are both z then we can conclude that the two variables are indeed equal to each other.

```
eq^Var z z = yes refl
```

Finally if the two variables are (s v) and (s w) respectively then we need to check recursively whether v is equal to w. If it is the case we can conclude by invoking the congruence rule for c. If v and w are not equal then a proof that (s v) and (s w) are will lead to a direct contradiction by injectivity of the constructor s.

```
eq^Var (s v) (s w) with eq^Var v w
... | yes p = yes (cong s p)
... | no ¬p = no λ where refl → ¬p refl
```

8.2.2 *Deciding Term Equality*

The constructor 'σ for descriptions gives us the ability to store values of any Set in terms. For some of these Sets (e.g. (ℕ → ℕ)), equality is not decidable. As a consequence our decision procedure will be conditioned to the satisfaction of a certain set of Constraints which we can compute from the Descr itself, as show in Figure 70. We demand that we are able to decide equality for all of the Sets mentioned in a description.

```
Constraints : Desc I → Set
Constraints ('σ A d)   = ((a b : A) → Dec (a ≡ b)) × (∀ a → Constraints (d a))
Constraints ('X _ _ d) = Constraints d
Constraints ('■ _)     = ⊤
```

Fig. 70. Constraints Necessary for Decidable Equality

Remembering that our descriptions are given a semantics as a big right-nested product terminated by an equality constraint, we realise that proving decidable equality will entail proving equality between proofs of equality. We are happy to assume Streicher's axiom K (Hofmann & Streicher (1994)) to easily dismiss this case. A more conservative approach would be to demand that equality is decidable on the index type I and to then use the classic Hedberg construction (Hedberg (1998)) to recover uniqueness of identity proofs for I.

Assuming that the constraints computed by (Constraints d) are satisfied, we define the decision procedure for equality of terms together with its equivalent for bodies. The function eq^Tm is a straightforward case analysis dismissing trivially impossible cases where terms have distinct head constructors ('var vs. 'con) and using either eq^Var or eq^⟦⟧ otherwise. The latter is defined by induction over e. The somewhat verbose definitions are not enlightening so we leave them out here.

We now have an informative decision procedure for equality between terms provided that the syntax they belong to satisfies a set of constraints. Other generic generic functions

```
eq^Tm : (t u : Tm d i σ Γ) → Dec (t ≡ u)
eq^⟦⟧ : ∀ e → Constraints e → (b c : ⟦ e ⟧ (Scope (Tm d i)) σ Γ) → Dec (b ≡ c)
```

Fig. 71. Type of Decidable Equality for Terms and Bodies

and decision procedure can be defined following the same approach: implement a similar function for variables first, compute a set of constraints, and demonstrate that they are sufficient to handle any input term.

9 Building Generic Proofs about Generic Programs

ACMM (2017) have already shown that, for the simply typed λ-calculus, introducing an abstract notion of Semantics not only reveals the shared structure of common traversals, it also allows them to give abstract proof frameworks for simulation or fusion lemmas. Their idea naturally extends to our generic presentation of semantics for all syntaxes.

The most important concept in this section is the *relator* (⟦ d ⟧^{R}), a relation transformer which characterises structurally equal layers such that their substructures are themselves related by the relation it is passed as an argument. It inherits a lot of its relational arguments' properties: whenever R is reflexive (respectively symmetric or transitive) so is (⟦ d ⟧^{R} R).

It is defined by induction on the description and case analysis on the two layers which are meant to be equal:

- In the stop token case '■ i, the two layers are considered to be trivially equal (i.e. the constraint generated is the unit type)
- When facing a recursive position 'X Δ j d, we demand that the two substructures are related by R Δ j and that the rest of the layers are related by (⟦ d ⟧^{R} R)
- Two nodes of type 'σ A d will be related if they both carry the same payload a of type A and if the rest of the layers are related by (⟦ d a ⟧^{R} R)

```
⟦_⟧^R : (d : Desc I) → (∀ Δ σ → ∀[ X Δ σ ⇒ Y Δ σ ⇒ const Set ])
                      → ∀[ ⟦ d ⟧ X σ ⇒ ⟦ d ⟧ Y σ ⇒ const Set ]
⟦ '■ j     ⟧^R R x       y        = ⊤
⟦ 'X Δ j d ⟧^R R (r , x) (r' , y) = R Δ j r r' × ⟦ d ⟧^R R x y
⟦ 'σ A d   ⟧^R R (a , x) (a' , y) = Σ (a' ≡ a) λ where refl → ⟦ d a ⟧^R R x y
```

Fig. 72. Zip: Characterising Structurally Equal Values with Related Substructures

If we were to take a fixpoint of ⟦_⟧^{R}, we could obtain a structural notion of equality for terms which we could prove equivalent to propositional equality. Although interesting in its own right, this section will focus on more advanced use-cases.

9.1 Simulation Lemma

A constraint mentioning the relator appears naturally when we want to say that a semantics can simulate another one. Given a relation $\mathcal{V}^R$ connecting values in $\mathcal{V}^A$ and $\mathcal{V}^B$, and a relation C^R connecting computations in C^A and C^B, we can define Kripke^R relating values Kripke $\mathcal{V}^A$ C^A and Kripke $\mathcal{V}^B$ C^B by stating that they send related inputs to related outputs. We use the relation transformer ∀[_] which lifts a relation on values to one on environments in a pointwise manner.

```
Kripke^R : ∀ Δ i → ∀[ Kripke 𝒱^A C^A Δ i ⇒ Kripke 𝒱^B C^B Δ i ⇒ const Set ]
Kripke^R []            σ k^A k^B = rel C^R σ k^A k^B
Kripke^R Δ@(_ :: _) σ k^A k^B = ∀ {Θ} (ρ : Thinning _ Θ) {vs^A vs^B} →
                                  All 𝒱^R Δ vs^A vs^B → rel C^R σ (k^A ρ vs^A) (k^B ρ vs^B)
```

Fig. 73. Relational Kripke Function Spaces: From Related Inputs to Related Outputs

We can then combine ⟦_⟧^{R} and Kripke^R to express the idea that two semantic objects of respective types ⟦ d ⟧ (Kripke $\mathcal{V}^A$ C^A) and ⟦ d ⟧ (Kripke $\mathcal{V}^B$ C^B) are synchronised. The simulation constraint on the algebras for two Semantics then becomes: given synchronized objects, the algebras should yield related computations. Together with self-explanatory constraints on var and th^$\mathcal{V}$, this constitutes the whole Simulation constraint, as show in Figure 74.

```
record Simulation (d : Desc I)
  (𝒮^A : Semantics d 𝒱^A C^A) (𝒮^B : Semantics d 𝒱^B C^B)
  (𝒱^R : Rel 𝒱^A 𝒱^B) (C^R : Rel C^A C^B) : Set where
  module 𝒮^A = Semantics 𝒮^A
  module 𝒮^B = Semantics 𝒮^B
  field th^R   : (ρ : Thinning Γ Δ) → rel 𝒱^R σ v^A v^B →
                 rel 𝒱^R σ (𝒮^A.th^𝒱 v^A ρ) (𝒮^B.th^𝒱 v^B ρ)
        var^R : rel 𝒱^R σ v^A v^B → rel C^R σ (𝒮^A.var v^A) (𝒮^B.var v^B)
        alg^R : (b : ⟦ d ⟧ (Scope (Tm d s)) σ Γ) → All 𝒱^R Γ ρ^A ρ^B →
                 let v^A = fmap d (𝒮^A.body ρ^A) b
                     v^B = fmap d (𝒮^B.body ρ^B) b
                 in ⟦ d ⟧^R (Kripke^R 𝒱^R C^R) v^A v^B → rel C^R σ (𝒮^A.alg v^A) (𝒮^B.alg v^B)
```

Fig. 74. A Generic Notion of Simulation

The fundamental lemma of simulations is a generic theorem showing that for each pair of Semantics respecting the Simulation constraint, we get related computations given environments of related input values. This theorem is once more mutually proven with a statement about Scopes, and Sizes play a crucial role in ensuring that the function is indeed total.

Instantiating this generic simulation lemma, we can for instance get that renaming and substitution are extensional (given extensionally equal environments they produce syntactically equal terms), or that renaming is a special case of substitution. Of course these results are not new but having them generically over all syntaxes with binding is convenient; which we have experienced first hand when tackling the POPLMark Reloaded challenge where rensub was actually needed.

```
sim  : All 𝒱ᴿ Γ ρᴬ ρᴮ → (t : Tm d s σ Γ) →
       rel 𝒞ᴿ σ (𝒮ᴬ.semantics ρᴬ t) (𝒮ᴮ.semantics ρᴮ t)
body : All 𝒱ᴿ Γ ρᴬ ρᴮ → ∀ Δ j → (t : Scope (Tm d s) Δ j Γ) →
       Kripkeᴿ 𝒱ᴿ 𝒞ᴿ Δ j (𝒮ᴬ.body ρᴬ Δ j t) (𝒮ᴮ.body ρᴮ Δ j t)

sim ρᴿ (‘var k) = varᴿ (lookupᴿ ρᴿ k)
sim ρᴿ (‘con t) = algᴿ t ρᴿ (liftᴿ d (body ρᴿ) t)

body ρᴿ []        i t = sim ρᴿ t
body ρᴿ (_ :: _) i t = λ σ vsᴿ → sim (vsᴿ >>ᴿ (thᴿ σ <$>ᴿ ρᴿ)) t
```

Fig. 75. Fundamental Lemma of Simulations

```
RenSub : Simulation d Ren Sub VarTmᴿ Eqᴿ

rensub : (ρ : Thinning Γ Δ) (t : Tm d ∞ σ Γ) → ren ρ t ≡ sub (‘var <$> ρ) t
rensub ρ = Simulation.sim RenSub (packᴿ λ _ → refl)
```

Fig. 76. Renaming as a Substitution via Simulation

When studying specific languages, new opportunities to deploy the fundamental lemma of simulations arise. Our solution to the POPLMark Reloaded challenge for instance describes the fact that sub ρ t reduces to sub ρ' t whenever for all v, $\rho(v)$ reduces to $\rho'(v)$ as a Simulation. The main theorem (strong normalisation of STLC via a logical relation) is itself an instance of (the unary version of) the simulation lemma.

The Simulation proof framework is the simplest example of the abstract proof frameworks ACMM (2017) introduce. They also explain how a similar framework can be defined for fusion lemmas and deploy it for the renaming-substitution interactions but also their respective interactions with normalisation by evaluation. Now that we are familiarised with the techniques at hand, we can tackle this more complex example for all syntaxes definable in our framework.

9.2 Fusion Lemma

Results which can be reformulated as the ability to fuse two traversals obtained as Semantics into one abound. When claiming that Tm is a Functor, we have to prove that two successive renamings can be fused into a single renaming where the Thinnings have been composed. Similarly, demonstrating that Tm is a relative Monad (Altenkirch *et al.* (2014)) implies proving that two consecutive substitutions can be merged into a single one whose environment is the first one, where the second one has been applied in a pointwise manner. The *Substitution Lemma* central to most model constructions (see for instance (Mitchell & Moggi (1991))) states that a syntactic substitution followed by the evaluation of the resulting term into the model is equivalent to the evaluation of the original term with an environment corresponding to the evaluated substitution.

A direct application of these results is our (to be published) entry to the POPLMark Reloaded challenge (2017). By using a Desc-based representation of intrinsically well typed and well scoped terms we directly inherit not only renaming and substitution but also all four fusion lemmas as corollaries of our generic results. This allows us to remove the usual boilerplate and go straight to the point. As all of these statements have precisely the same structure, we can once more devise a framework which will, provided that its constraints are satisfied, prove a generic fusion lemma.

Fusion is more involved than simulation so we will step through each one of the constraints individually, trying to give the reader an intuition for why they are shaped the way they are.

9.2.1 The Fusion Constraints

The notion of fusion is defined for a triple of Semantics; each $\mathcal{S}^i$ being defined for values in $\mathcal{V}^i$ and computations in $\mathcal{C}^i$. The fundamental lemma associated to such a set of constraints will state that running $\mathcal{S}^B$ after $\mathcal{S}^A$ is equivalent to running $\mathcal{S}^{AB}$ only.

The definition of fusion is parametrised by three relations: $\mathcal{E}^R$ relates triples of environments of values in (Γ –Env) $\mathcal{V}^A$ Δ, (Δ –Env) $\mathcal{V}^B$ Θ and (Γ –Env) $\mathcal{V}^{AB}$ Θ respectively; $\mathcal{V}^R$ relates pairs of values $\mathcal{V}^B$ and $\mathcal{V}^{AB}$; and $\mathcal{C}^R$, our notion of equivalence for evaluation results, relates pairs of computation in $\mathcal{C}^B$ and $\mathcal{C}^{AB}$.

```
record Fusion (d : Desc I) (𝒮ᴬ : Semantics d 𝒱ᴬ 𝒞ᴬ) (𝒮ᴮ : Semantics d 𝒱ᴮ 𝒞ᴮ)
  (𝒮ᴬᴮ : Semantics d 𝒱ᴬᴮ 𝒞ᴬᴮ)
  (ℰᴿ : ∀ Γ Δ {Θ} → (Γ –Env) 𝒱ᴬ Δ → (Δ –Env) 𝒱ᴮ Θ → (Γ –Env) 𝒱ᴬᴮ Θ → Set)
  (𝒱ᴿ : Rel 𝒱ᴮ 𝒱ᴬᴮ) (𝒞ᴿ : Rel 𝒞ᴮ 𝒞ᴬᴮ) : Set where
```

The first obstacle we face is the formal definition of “running $\mathcal{S}^B$ after $\mathcal{S}^A$”: for this statement to make sense, the result of running $\mathcal{S}^A$ ought to be a term. Or rather, we ought to be able to extract a term from a $\mathcal{C}^A$. Hence the first constraint: the existence of a reify^A function, which we supply as a field of the record Fusion. When dealing with syntactic semantics such as renaming or substitution this function will be the identity. However nothing prevents to try to prove for instance that normalisation by evaluation is idempotent in which case a bona fide reification function extracting terms from model values will be used.

```
reifyᴬ : ∀ σ → ∀[ 𝒞ᴬ σ ⇒ Tm d ∞ σ ]
```

Then, we have to think about what happens when going under a binder: $\mathcal{S}^A$ will produce a Kripke function space where a syntactic value is required. Provided that $\mathcal{V}^A$ is VarLike, we can make use of reify to get a Scope back. Hence the second constraint.

```
vl^𝒱ᴬ : VarLike 𝒱ᴬ
```

Still thinking about going under binders: if three evaluation environments ρ^A in (Γ –Env) $\mathcal{V}^A$ Δ, ρ^B in (Δ –Env) $\mathcal{V}^B$ Θ, and ρ^{AB} in (Γ –Env) $\mathcal{V}^{AB}$ Θ are related by $\mathcal{E}^R$ and we are given a thinning σ from Θ to Ω then ρ^A, the thinned ρ^B and the thinned ρ^{AB} should still be related.

$$\text{th}^\wedge\mathcal{E}^R : \mathcal{E}^R\ \Gamma\ \Delta\ \rho^A\ \rho^B\ \rho^{AB} \to (\rho : \text{Thinning}\ \Theta\ \Omega) \to \\ \quad \mathcal{E}^R\ \Gamma\ \Delta\ \rho^A\ (\text{th}^\wedge\text{Env}\ \mathcal{S}^B.\text{th}^\wedge\mathcal{V}\ \rho^B\ \rho)\ (\text{th}^\wedge\text{Env}\ \mathcal{S}^{AB}.\text{th}^\wedge\mathcal{V}\ \rho^{AB}\ \rho)$$

Remembering that _>>_ is used in the definition of body (Figure 28) to combine two disjoint environments $(\Gamma\ \text{–Env})\ \mathcal{V}\ \Theta$ and $(\Delta\ \text{–Env})\ \mathcal{V}\ \Theta$ into one of type $((\Gamma +\!\!+ \Delta)\ \text{–Env})\ \mathcal{V}\ \Theta)$, we mechanically need a constraint stating that _>>_ is compatible with $\mathcal{E}^R$. We demand as an extra precondition that the values ρ^B and ρ^{AB} are extended with are related according to $\mathcal{V}^R$. Lastly, for all the types to match up, ρ^A has to be extended with placeholder variables.

$$\_{>>}^R\_ : \mathcal{E}^R\ \Gamma\ \Delta\ \rho^A\ \rho^B\ \rho^{AB} \to \text{All}\ \mathcal{V}^R\ \Theta\ vs^B\ vs^{AB} \to \\ \quad \text{let}\ id{>>}\rho^A = \text{fresh}^l\ \text{vl}^\wedge\mathcal{V}^A\ \Delta >> \text{th}^\wedge\text{Env}\ \mathcal{S}^A.\text{th}^\wedge\mathcal{V}\ \rho^A\ (\text{fresh}^r\ \text{vl}^\wedge\text{Var}\ \Theta) \\ \quad \text{in}\ \mathcal{E}^R\ (\Theta +\!\!+ \Gamma)\ (\Theta +\!\!+ \Delta)\ id{>>}\rho^A\ (vs^B >> \rho^B)\ (vs^{AB} >> \rho^{AB})$$

We finally arrive at the constraints focusing on the semantical counterparts of the terms' constructors. Each constraint essentially states that evaluating a term with $\mathcal{S}^A$, reifying the result and running $\mathcal{S}^B$ is equivalent to using $\mathcal{S}^{AB}$ straight away. This can be made formal by defining the following relation $\mathcal{R}$.

$$\mathcal{R} : \forall\ \sigma \to (\Gamma\ \text{–Env})\ \mathcal{V}^A\ \Delta \to (\Delta\ \text{–Env})\ \mathcal{V}^B\ \Theta \to (\Gamma\ \text{–Env})\ \mathcal{V}^{AB}\ \Theta \to \\ \quad \text{Tm}\ d\ s\ \sigma\ \Gamma \to \text{Set} \\ \mathcal{R}\ \sigma\ \rho^A\ \rho^B\ \rho^{AB}\ t = \text{rel}\ \mathcal{C}^R\ \sigma\ (\text{eval}^B\ \rho^B\ (\text{reify}^A\ \sigma\ (\text{eval}^A\ \rho^A\ t)))\ (\text{eval}^{AB}\ \rho^{AB}\ t)$$

When evaluating a variable, on the one hand $\mathcal{S}^A$ will look up its meaning in the evaluation environment, turn the resulting value into a computation which will get reified and then the result will be evaluated with $\mathcal{S}^B$. Provided that all three evaluation environments are related by $\mathcal{E}^R$ this should be equivalent to looking up the value in $\mathcal{S}^{AB}$'s environment and turning it into a computation. Hence the constraint var^R:

$$\text{var}^R : \mathcal{E}^R\ \Gamma\ \Delta\ \rho^A\ \rho^B\ \rho^{AB} \to \forall\ v \to \mathcal{R}\ \sigma\ \rho^A\ \rho^B\ \rho^{AB}\ (\text{`var}\ v)$$

The case of the algebra follows a similar idea albeit being more complex: a term gets evaluated using $\mathcal{S}^A$ and to be able to run $\mathcal{S}^B$ afterwards we need to recover a piece of syntax. This is possible if the Kripke functional spaces are reified by being fed placeholder $\mathcal{V}^A$ arguments (which can be manufactured thanks to the $\text{vl}^\wedge\mathcal{V}^A$ we mentioned before) and then quoted. Provided that the result of running $\mathcal{S}^B$ on that term is related via $[\![\ d\ ]\!]^R\ (\text{Kripke}^R\ \mathcal{V}^R\ \mathcal{C}^R)$ to the result of running $\mathcal{S}^{AB}$ on the original term, the alg^R constraint states that the two evaluations yield related computations.

$$\text{alg}^R : \mathcal{E}^R\ \Gamma\ \Delta\ \rho^A\ \rho^B\ \rho^{AB} \to (b : [\![\ d\ ]\!]\ (\text{Scope}\ (\text{Tm}\ d\ s))\ \sigma\ \Gamma) \to \\ \quad \text{let}\ b^A : [\![\ d\ ]\!]\ (\text{Kripke}\ \mathcal{V}^A\ \mathcal{C}^A)\ \_\ \_ \\ \quad\quad b^A = \text{fmap}\ d\ (\mathcal{S}^A.\text{body}\ \rho^A)\ b \\ \quad\quad b^B = \text{fmap}\ d\ (\lambda\ \Delta\ i \to \mathcal{S}^B.\text{body}\ \rho^B\ \Delta\ i \circ \text{quote}^A\ \Delta\ i)\ b^A \\ \quad\quad b^{AB} = \text{fmap}\ d\ (\mathcal{S}^{AB}.\text{body}\ \rho^{AB})\ b \\ \quad \text{in}\ [\![\ d\ ]\!]^R\ (\text{Kripke}^R\ \mathcal{V}^R\ \mathcal{C}^R)\ b^B\ b^{AB} \to \mathcal{R}\ \sigma\ \rho^A\ \rho^B\ \rho^{AB}\ (\text{`con}\ b)$$

9.2.2 The Fundamental Lemma of Fusion

This set of constraint is enough to prove a fundamental lemma of Fusion stating that from a triple of related environments, one gets a pair of related computations: the composition of

$\mathcal{S}^A$ and $\mathcal{S}^B$ on one hand and $\mathcal{S}^{AB}$ on the other. This lemma is once again proven mutually with its counterpart for Semantics's body's action on Scopes.

```
fusion : ℰᴿ Γ Δ ρᴬ ρᴮ ρᴬᴮ → (t : Tm d s σ Γ) → ℛ σ ρᴬ ρᴮ ρᴬᴮ t
```

Fig. 77. Fundamental Lemma of Fusion

9.2.3 *Instances of Fusion*

A direct consequence of this result is the four lemmas collectively stating that any pair of renamings and / or substitutions can be fused together to produce either a renaming (in the renaming-renaming interaction case) or a substitution (in all the other cases). One such example is the fusion of substitution followed by renaming into a single substitution where the renaming has been applied to the environment.

```
subren : (t : Tm d i σ Γ) (ρ₁ : (Γ −Env) (Tm d ∞) Δ) (ρ₂ : Thinning Δ Θ) →
         ren ρ₂ (sub ρ₁ t) ≡ sub (ren ρ₂ <$> ρ₁) t
```

Fig. 78. A Corollary: Substitution-Renaming Fusion

Another corollary of the fundamental lemma of fusion is the observation that Kaiser, Schäfer, and Stark (2018) make: *assuming functional extensionality*, all the ACMM (2017) traversals are compatible with variable renaming. We can reproduce this result generically for all syntaxes (see accompanying code) but refrain from using it in practice when an axiom-free alternative is provable. Their observation naturally raises the question of whether the same semantics are also stable under substitution. Our semantics implementing printing with names is a clear counter-example.

9.3 Definition of Bisimilarity for Co-finite Objects

Although we were able to use propositional equality when studying syntactic traversals working on terms, it is not the appropriate notion of equality for co-finite trees. What we want is a generic coinductive notion of bisimilarity for all co-finite tree types obtained as the unfolding of a description. Two trees are bisimilar if their top layers have the same shape and their substructures are themselves bisimilar. This is precisely the type of relation $⟦\_⟧^R$ was defined to express. Hence the following coinductive relation.

```
record ≈^∞Tm (d : Desc I) (s : Size) (i : I) (t u : ∞Tm d s i) : Set where
  coinductive
  field force : {s′ : Size< s} → ⟦ d ⟧ᴿ (λ _ i → ≈^∞Tm d s′ i) (t .force) (u .force)
```

Fig. 79. Generic Notion of Bisimilarity for Co-finite Trees

We can then prove by coinduction that this generic definition always gives rise to an equivalence relation by using the relator's stability properties (if R is reflexive / symmetric / transitive then so is ($\llbracket d \rrbracket^R R$) mentioned in Section 9.

```
refl  : ≈^∞Tm d s i t t
sym   : ≈^∞Tm d s i t u → ≈^∞Tm d s i u t
trans : ≈^∞Tm d s i t u → ≈^∞Tm d s i u v → ≈^∞Tm d s i t v
```

This definition can be readily deployed to prove e.g. that the unfolding of 01↺ defined in Section 8.1 is indeed bisimilar to 01··· which was defined in direct style. The proof is straightforward due to the simplicity of this example: the first refl witnesses the fact that both definitions pick the same constructor (a cons cell), the second that they carry the same natural number, and we can conclude by an appeal to the coinduction hypothesis.

```
eq-01 : {i : Size} → ≈^∞Tm (CListD ℕ) i tt 01··· (unfold 01↺)
eq-10 : {i : Size} → ≈^∞Tm (CListD ℕ) i tt 10··· (unfold (1 :: 0 :: 1 :: ↶ s z))

eq-01 .force = refl , refl , eq-10 , tt
eq-10 .force = refl , refl , eq-01 , tt
```

10 Related Work

10.1 Variable Binding

The representation of variable binding in formal systems has been a hot topic for decades. Part of the purpose of the first POPLMark challenge (2005) was to explore and compare various methods.

Having based our work on a de Bruijn encoding of variables, and thus a canonical treatment of α-equivalence classes, our work has no direct comparison with permutation-based treatments such as those of Pitts' and Gabbay's nominal syntax (2002).

Our generic universe of syntax is based on scoped and typed de Bruijn indices (de Bruijn (1972)) but it is not a necessity. It is for instance possible to give an interpretation of Descriptions corresponding to Chlipala's Parametric Higher-Order Abstract Syntax (2008) and we would be interested to see what the appropriate notion of Semantics is for this representation.

10.2 Alternative Binding Structures

The binding structure we present here is based on a flat, lexical scoping strategy. There are other strategies and it would be interesting to see whether our approach could be reused in these cases.

Weirich, Yorgey, and Sheard's work (2011) encompassing a large array of patterns (nested, recursive, telescopic, and n-ary) can inform our design. They do not enforce scoping invariants internally which forces them to introduce separate constructors for a simple binder, a recursive one, or a telescopic pattern. They recover guarantees by giving their syntaxes a nominal semantics thus bolting down the precise meaning of each combinators and then proving that users may only generate well formed terms.

Bach Poulsen, Rouvoet, Tolmach, Krebbers and Visser (2018) introduce notions of scope graphs and frames to scale the techniques typical of well scoped and typed deep embeddings to imperative languages. They can already handle a large subset of Middleweight Java.

We have demonstrated how to write generic programs over the potentially cyclic structures of Ghani, Hamana, Uustalu and Vene (2006). Further work by Hamana (2009) yielded a different presentation of cyclic structures which preserves sharing: pointers can not only refer to nodes above them but also across from them in the cyclic tree. Capturing this class of inductive types as a set of syntaxes with binding and writing generic programs over them is still an open problem.

10.3 Semantics of Syntaxes with Binding

An early foundational study of a general *semantic* framework for signatures with binding, algebras for such signatures, and initiality of the term algebra, giving rise to a categorical 'program' for substitution and proofs of its properties, was given by Fiore, Plotkin and Turi (Fiore *et al.* (1999)), working in the category of presheaves over renamings, (a skeleton of) the category of finite sets. The presheaf condition corresponds to our notion of being Thinnable. Exhibiting algebras based on both de Bruijn *level* and *index* encodings, their approach isolates the usual (abstract) arithmetic required of such encodings.

By contrast, working in an *implemented* type theory, where the encoding can be understood as its own foundation, without appeal to an external mathematical semantics, we are able to go further in developing machine-checked such implementations and proofs, themselves generic with respect to an abstract syntax Desc of syntaxes-with-binding. Moreover, the usual source of implementation anxiety, namely concrete arithmetic on de Bruijn indices, has been successfully encapsulated via the □ coalgebra structure. It is perhaps noteworthy that our type-theoretic constructions, by contrast with their categorical ones, appear to make fewer commitments as to functoriality, thinnability, etc. in our specification of semantics, with such properties typically being *provable* as a further instance of our framework.

10.4 Meta-Theory Automation via Tactics and Code Generation

The tediousness of repeatedly proving similar statements has unsurprisingly led to various attempts at automating the pain away via either code generation or the definition of tactics. These solutions can be seen as untrusted oracles driving the interactive theorem prover.

Polonowski's DBGen (2013) takes as input a raw syntax with comments annotating binding sites. It generates a module defining lifting, substitution as well as a raw syntax using names and a validation function transforming named terms into de Bruijn ones; we refrain from calling it a scopechecker as terms are not statically proven to be well scoped.

Kaiser, Schäfer, and Stark (2018) build on our previous paper to draft possible theoretical foundations for Autosubst, a so-far untrusted set of tactics. The paper is based on a specific syntax: well-scoped call-by-value System F. In contrast, our effort has been here to carve out a precise universe of syntaxes with binding and give a systematic account of their semantics and proofs.

Keuchel, Weirich, and Schrijvers' Needle (2016) is a code generator written in Haskell producing syntax-specific Coq modules implementing common traversals and lemmas about them.

10.5 Universes of Syntaxes with Binding

Keeping in mind Altenkirch and McBride's observation that generic programming is everyday programming in dependently-typed languages (2002), we can naturally expect generic, provably sound, treatments of these notions in tools such as Agda or Coq.

Keuchel (2011) together with Jeuring (2012) define a universe of syntaxes with binding with a rich notion of binding patterns closed under product but also sum as long as the disjoint patterns bind the same variables. They give their universe two distinct semantics: a first one based on well scoped de Bruijn indices and a second one based on Parametric Higher-Order Abstract Syntax (PHOAS) (Chlipala (2008)) together with a generic conversion function from the de Bruijn syntax to the PHOAS one. Following McBride (2005), they implement both renaming and substitution in one swift move. They leave other opportunities for generic programming and proving to future work.

Keuchel, Weirich, and Schrijvers' Knot (2016) implements as a set of generic programs the traversals and lemmas generated in specialised forms by their Needle program. They see Needle as a pragmatic choice: working directly with the free monadic terms over finitary containers would be too cumbersome. In our experience solving the POPLMark Reloaded challenge, Agda's pattern synonyms (Pickering *et al.* (2016)) make working with an encoded definition almost seamless.

The GMeta generic framework (2012) provides a universe of syntaxes and offers various binding conventions (locally nameless (Charguéraud (2012)) or de Bruijn indices). It also generically implements common traversals (e.g. computing the sets of free variables, shifting de Bruijn indices or substituting terms for parameters) as well as common predicates (e.g. being a closed term) and provides generic lemmas proving that they are well behaved. It does not offer a generic framework for defining new well scoped-and-typed semantics and proving their properties.

Érdi (2018) defines a universe inspired by a first draft of this paper and gives three different interpretations (raw, scoped and typed syntax) related via erasure. He provides scope- and type- preserving renaming and substitution as well as various generic proofs that they are well behaved but offers neither a generic notion of semantics, nor generic proof frameworks.

Copello (2017) works with *named* binders and defines nominal techniques (e.g. name swapping) and ultimately α-equivalence over a universe of regular trees with binders inspired by Morris' (2006).

10.6 Fusion of Successive Traversals

The careful characterisation of the successive recursive traversals which can be fused together into a single pass in a semantics-preserving way is not new. This transformation is a much needed optimisation principle in a high-level functional language.

Through the careful study of the recursion operator associated to each strictly positive datatype, Malcolm (1990) defined optimising fusion proof principles. Other optimisations such as deforestation (Wadler (1990)) or the compilation of a recursive definition into an equivalent abstract machine-based tail-recursive program (Cortiñas & Swierstra (2018)) rely on similar generic proofs that these transformations are meaning-preserving.

11 Conclusion and Future Work

Recalling Allais, Chapman, McBride and McKinna's earlier work (2017) we have started from an example of a scope- and type- safe language (the simply typed λ-calculus), have studied common invariant preserving traversals and noticed their similarity. After introducing a notion of semantics and refactoring these traversals as instances of the same fundamental lemma, we have observed the tight connection between the abstract definition of semantics and the shape of the language.

By extending a universe of datatype descriptions to support a notion of binding, we have given a generic presentation of syntaxes with binding. We then described a large class of scope- and type- safe generic programs acting on all of them We started with syntactic traversals such as renaming and substitution. We then demonstrated how to write a small compiler pipeline: scope checking, type checking and elaboration to a core language, desugaring of new constructors added by a language transformer, dead code elimination and inlining, partial evaluation, and printing with names.

We have seen how to construct generic proofs about these generic programs. We first introduced a Simulation relation showing what it means for two semantics to yield related outputs whenever they are fed related input environments. We then built on our experience to tackle a more involved case: identifying a set of constraints guaranteeing that two semantics run consecutively can be subsumed by a single pass of a third one.

We have put all of these results into practice by using them to solve the (to be published) POPLMark Reloaded challenge which consists of formalising strong normalisation for the simply typed λ-calculus via a logical-relation argument. This also gave us the opportunity to try our framework on larger languages by tackling the challenge's extensions to sum types and Gödel's System T.

Finally, we have demonstrated that this formalisation can be re-used in other domains by seeing our syntaxes with binding as potentially cyclic terms. Their unfolding is a non-standard semantics and we provide the user with a generic notion of bisimilarity to reason about them.

11.1 Limitations of the Current Framework

Although quite versatile already our current framework has some limitations which suggest avenues for future work. We list these limitations from easiest to hardest to resolve. Remember that each modification to the universe of syntaxes needs to be given an appropriate semantics.

Closure under Products Our current universe of descriptions is closed under sums as demonstrated in Section 5. It is however not closed under products: two arbitrary right-nested products conforming to a description may disagree on the kind of the term they are constructing. An approach where the kind is an input from which the description of allowed constructors is computed (à la Dagand (2013) where, for instance, the 'lam constructor is only offered if the input kind is a function type) would not suffer from this limitation.

Unrestricted Variables Our current notion of variable can be used to form a term of any kind. We remarked in Sections 7.3 and 7.4 that in some languages we want to restrict this ability to one kind in particular. In that case, we wanted users to only be able to use variables at the kind Infer of our bidirectional language. For the time being we made do by restricting the environment values our Semantics use to a subset of the kinds: terms with variables of the wrong kind will not be given a semantics.

Flat Binding Structure Our current setup limits us to flat binding structures: variable and binder share the same kinds. This prevents us from representing languages with binding patterns, for instance pattern-matching let-binders which can have arbitrarily nested patterns taking pairs apart.

Closure under Derivation One-hole contexts play a major role in the theory of programming languages. Just like the one-hole context of a datatype is a datatype (Abbott *et al.* (2005)), we would like our universe to be closed under derivatives so that the formalisation of e.g. evaluation contexts could benefit directly from the existing machinery.

Closure under Closures Jander's work on formalising and certifying continuation passing style transformations (Jander (2019)) highlighted the need for a notion of syntaxes with closures. Recalling that our notion of Semantics is always compatible with a preliminary renaming (Kaiser *et al.* (2018)) but not necessarily a preliminary substitution (printing is for instance not stable under substitution), accommodating terms with suspended substitutions is a real challenge. Preliminary experiments show that a drastic modification of the type of the fundamental lemma of Semantics makes dealing with such closures possible. Whether the resulting traversal has good properties that can be proven generically is still an open problem.

11.2 Future work

The diverse influences leading to this work suggest many opportunities for future research.

- Our example of the elaboration of an enriched language to a core one, ACMM's implementation of a Continuation Passing Style conversion function, and Jander's work (2019) on the certification of a intrinsically typed CPS transformation raises the question of how many such common compilation passes can be implemented generically.

- Our universe only includes syntaxes that allow unrestricted variable use. Variables may be used multiple times or never, with no restriction. We are interested in representing syntaxes that only allow single use of variables, such as term calculi for linear logic (Benton *et al.* (1993); Barber (1996); Atkey & Wood (2018)), or that annotate variables with usage information (Brunel *et al.* (2014); Ghica & Smith (2014); Petricek *et al.* (2014)), or arrange variables into non-list like structures such as bunches (O'Hearn (2003)), or arbitrary algebraic structures (Licata *et al.* (2017)), and in investigating what form a generic semantics for these syntaxes takes.
- An extension of McBride's theory of ornaments (2017) could provide an appropriate framework to formalise and mechanise the connection between various languages, some being seen as refinements of others. This is particularly evident when considering the informative typechecker (see the accompanying code) which given a scoped term produces a scoped-and-typed term by type-checking or type-inference.
- Our work on the POPLMark Reloaded challenge highlights a need for generic notions of congruence closure which would come with guarantees (if the original relation is stable under renaming and substitution so should the closure). Similarly, the "evaluation contexts" corresponding to a syntax could be derived automatically by building on the work of Huet (1997) and Abbott, Altenkirch, McBride and Ghani (2005), allowing us to revisit previous work based on concrete instances of ACMM such as McLaughlin, McKinna and Stark (2018).

We now know how to generically describe syntaxes and their well behaved semantics. We can now start asking what it means to define well behaved judgments. Why stop at helping the user write their specific language's meta-theory when we could study meta-meta-theory?

References

Abbott, Michael Gordon, Altenkirch, Thorsten, McBride, Conor, & Ghani, Neil. (2005). δ for data: Differentiating data structures. *Fundam. inform.*, **65**(1-2), 1–28.

Abel, Andreas. (2010). Miniagda: Integrating sized and dependent types. *Pages 14–28 of:* Bove, Ana, Komendantskaya, Ekaterina, & Niqui, Milad (eds), *Proceedings Workshop on Partiality and Recursion in Interactive Theorem Provers, PAR 2010, Edinburgh, UK, 15th July 2010.* EPTCS, vol. 43.

Abel, Andreas, Pientka, Brigitte, Thibodeau, David, & Setzer, Anton. (2013). Copatterns: programming infinite structures by observations. *Pages 27–38 of: ACM SIGPLAN Notices*, vol. 48. ACM.

Abel, Andreas, Momigliano, Alberto, & Pientka, Brigitte. (2017). Poplmark reloaded. *Proceedings of the Logical Frameworks and Meta-Languages: Theory and Practice workshop*.

Allais, Guillaume. (2018). agdarsec – Total parser combinators. *Pages 45–59 of:* Boldo, Sylvie, & Magaud, Nicolas (eds), *JFLA 2018 Journées Francophones des Langages Applicatifs*. Banyuls-sur-Mer, France: publié par les auteurs.

Allais, Guillaume, Chapman, James, McBride, Conor, & McKinna, James. (2017). Type-and-scope safe programs and their proofs. *Pages 195–207 of: Proceedings of the 6th ACM SIGPLAN Conference on Certified Programs and Proofs*. CPP 2017. ACM.

Altenkirch, Thorsten, & McBride, Conor. (2002). Generic programming within dependently typed programming. *Pages 1–20 of:* Gibbons, Jeremy, & Jeuring, Johan (eds), *Generic Programming, IFIP TC2/WG2.1 Working Conference on Generic Programming, July 11-12, 2002, Dagstuhl, Germany*. IFIP Conference Proceedings, vol. 243. Kluwer.

Altenkirch, Thorsten, & Reus, Bernhard. (1999). Monadic presentations of lambda terms using generalized inductive types. *Pages 453–468 of: CSL*. Springer.

Altenkirch, Thorsten, Hofmann, Martin, & Streicher, Thomas. (1995). Categorical reconstruction of a reduction free normalization proof. *Pages 182–199 of: LNCS*, vol. 530. Springer.

Altenkirch, Thorsten, Chapman, James, & Uustalu, Tarmo. (2014). Relative monads formalised. *Journal of formalized reasoning*, **7**(1), 1–43.

Altenkirch, Thorsten, Chapman, James, & Uustalu, Tarmo. (2015). Monads need not be endofunctors. *Logical methods in computer science*, **11**(1).

Appel, Andrew W., & Jim, Trevor. (1997). Shrinking lambda expressions in linear time. *J. funct. program.*, **7**(5), 515–540.

Atkey, Robert. (2015). *An algebraic approach to typechecking and elaboration*. Talk.

Atkey, Robert, & Wood, James. (2018). Context constrained computation. *3rd Workshop on Type-Driven Development (TyDe '18), Extended Abstract*.

Aydemir, Brian E., Bohannon, Aaron, Fairbairn, Matthew, Foster, J. Nathan, Pierce, Benjamin C., Sewell, Peter, Vytiniotis, Dimitrios, Washburn, Geoffrey, Weirich, Stephanie, & Zdancewic, Steve. (2005). Mechanized metatheory for the masses: The poplmark challenge. *Pages 50–65 of:* Hurd, Joe, & Melham, Tom (eds), *Theorem Proving in Higher Order Logics*. Springer.

Bach Poulsen, Casper, Rouvoet, Arjen, Tolmach, Andrew, Krebbers, Robbert, & Visser, Eelco. (2018). Intrinsically-typed definitional interpreters for imperative languages. *Proc. acm program. lang.*, **2**(POPL), 16:1–16:34.

Barber, Andrew. (1996). *Dual intuitionistic linear logic*. Tech. rept. ECS-LFCS-96-347. LFCS, University of Edinburgh.

Bellegarde, Françoise, & Hook, James. (1994). Substitution: A formal methods case study using monads and transformations. *Science of computer programming*, **23**(2), 287 – 311.

Benke, Marcin, Dybjer, Peter, & Jansson, Patrik. (2003). Universes for generic programs and proofs in dependent type theory. *Nordic j. of computing*, **10**(4), 265–289.

Benton, Nick, Hur, Chung-Kil, Kennedy, Andrew J, & McBride, Conor. (2012). Strongly typed term representations in Coq. *Jar*, **49**(2), 141–159.

Benton, P. N., Bierman, Gavin M., de Paiva, Valeria, & Hyland, Martin. (1993). A term calculus for intuitionistic linear logic. *Pages 75–90 of:* Bezem, Marc, & Groote, Jan Friso (eds), *Typed Lambda Calculi and Applications, International Conference on Typed Lambda Calculi and Applications, TLCA '93, Utrecht, The Netherlands, March 16-18, 1993, Proceedings*. Lecture Notes in Computer Science, vol. 664. Springer.

Bird, Richard S., & Paterson, Ross. (1999). de Bruijn notation as a nested datatype. *Journal of functional programming*, **9**(1), 77–91.

Brady, Edwin. (2013). Idris, a general-purpose dependently typed programming language: Design and implementation. *Journal of functional programming*, **23**(5), 552–593.

Brady, Edwin, & Hammond, Kevin. (2006). A verified staged interpreter is a verified compiler. *Pages 111–120 of:* Jarzabek, Stan, Schmidt, Douglas C., & Veldhuizen, Todd L. (eds), *Generative Programming and Component Engineering, 5th International Conference, GPCE 2006, Portland, Oregon, USA, October 22-26, 2006, Proceedings*. ACM.

Brunel, Aloïs, Gaboardi, Marco, Mazza, Damiano, & Zdancewic, Steve. (2014). A Core Quantitative Coeffect Calculus. *Pages 351–370 of: Programming Languages and Systems - 23rd European Symposium on Programming, ESOP 2014*.

Chapman, James, Dagand, Pierre-Évariste, McBride, Conor, & Morris, Peter. (2010). The gentle art of levitation. *Pages 3–14 of: Proceedings of the 15th ACM SIGPLAN International Conference on Functional Programming*. ICFP '10. ACM.

Chapman, James Maitland. (2009). *Type checking and normalisation*. Ph.D. thesis, University of Nottingham (UK).

Charguéraud, Arthur. (2012). The locally nameless representation. *Journal of automated reasoning*, **49**(3), 363–408.

Chlipala, Adam. (2008). Parametric higher-order abstract syntax for mechanized semantics. *Pages 143–156 of:* Hook, James, & Thiemann, Peter (eds), *Proceeding of the 13th ACM SIGPLAN international conference on Functional programming, ICFP 2008, Victoria, BC, Canada, September 20-28, 2008*. ACM.

Copello, Ernesto. (2017). *On the formalisation of the metatheory of the lambda calculus and languages with binders*. Ph.D. thesis, Universidad de la República (Uruguay).

Coquand, Catarina. (2002). A formalised proof of the soundness and completeness of a simply typed lambda-calculus with explicit substitutions. *Higher-order and symbolic computation*, **15**(1), 57–90.

Cortiñas, Carlos Tomé, & Swierstra, Wouter. (2018). From algebra to abstract machine: a verified generic construction. *Pages 78–90 of:* Eisenberg, Richard A., & Vazou, Niki (eds), *Proceedings of the 3rd ACM SIGPLAN International Workshop on Type-Driven Development, TyDe@ICFP 2018, St. Louis, MO, USA, September 27, 2018*. ACM.

Dagand, Pierre-Évariste. (2013). *A cosmology of datatypes : reusability and dependent types*. Ph.D. thesis, University of Strathclyde, Glasgow, UK.

Danielsson, Nils Anders. (2010). Total parser combinators. *Pages 285–296 of:* Hudak, Paul, & Weirich, Stephanie (eds), *Proceeding of the 15th ACM SIGPLAN international conference on Functional programming, ICFP 2010, Baltimore, Maryland, USA, September 27-29, 2010*. ACM.

de Bruijn, Nicolaas Govert. (1972). Lambda Calculus notation with nameless dummies. *Pages 381–392 of: Indagationes Mathematicae*, vol. 75. Elsevier.

de Moura, Leonardo Mendonça, Kong, Soonho, Avigad, Jeremy, van Doorn, Floris, & von Raumer, Jakob. (2015). The lean theorem prover (system description). *Pages 378–388 of:* Felty, Amy P., & Middeldorp, Aart (eds), *Automated Deduction - CADE-25 - 25th International Conference on Automated Deduction, Berlin, Germany, August 1-7, 2015, Proceedings*. Lecture Notes in Computer Science, vol. 9195. Springer.

Dybjer, Peter. (1994). Inductive families. *Formal aspects of computing*, **6**(4), 440–465.

Dybjer, Peter, & Setzer, Anton. (1999). A finite axiomatization of inductive-recursive definitions. *Pages 129–146 of:* Girard, Jean-Yves (ed), *Typed Lambda Calculi and Applications, 4th International Conference, TLCA'99, L'Aquila, Italy, April 7-9, 1999, Proceedings*. Lecture Notes in Computer Science, vol. 1581. Springer.

Eisenberg, Richard A. (2018). *Stitch: The sound type-indexed type checker*. Draft.

Érdi, Gergő. (2018). *Generic description of well-scoped, well-typed syntaxes*. Unpublished draft, privately communicated.

Fiore, Marcelo P., Plotkin, Gordon D., & Turi, Daniele. (1999). Abstract syntax and variable binding. *Pages 193–202 of: 14th Annual IEEE Symposium on Logic in Computer Science, Trento, Italy, July 2-5, 1999*. IEEE Computer Society.

Gabbay, Murdoch, & Pitts, Andrew M. (2002). A new approach to abstract syntax with variable binding. *Formal asp. comput.*, **13**(3-5), 341–363.

Ghica, Dan R., & Smith, Alex I. (2014). Bounded linear types in a resource semiring. *Pages 331–350 of: Programming Languages and Systems - 23rd European Symposium on Programming, ESOP 2014*.

Gibbons, Jeremy, & d. S. Oliveira, Bruno C. (2009). The essence of the Iterator pattern. *J. funct. program.*, **19**(3-4), 377–402.

Hamana, Makoto. (2009). Initial algebra semantics for cyclic sharing structures. *Pages 127–141 of:* Curien, Pierre-Louis (ed), *Typed Lambda Calculi and Applications, 9th International Conference, TLCA 2009, Brasilia, Brazil, July 1-3, 2009. Proceedings*. Lecture Notes in Computer Science, vol. 5608. Springer.

Hedberg, Michael. (1998). A coherence theorem for Martin-Löf's type theory. *J. funct. program.*, **8**(4), 413–436.

Hinze, Ralf, & Peyton Jones, Simon L. (2000). Derivable type classes. *Electr. notes theor. comput. sci.*, **41**(1), 5–35.

Hirschowitz, André, & Maggesi, Marco. (2012). Nested abstract syntax in coq. *J. autom. reasoning*, **49**(3), 409–426.

Hofmann, Martin, & Streicher, Thomas. (1994). The groupoid model refutes uniqueness of identity proofs. *Pages 208–212 of: Proceedings of the Ninth Annual Symposium on Logic in Computer Science (LICS '94), Paris, France, July 4-7, 1994*. IEEE Computer Society.

Hudak, Paul. (1996). Building domain-specific embedded languages. *Acm computing surveys (csur)*, **28**(4es), 196.

Huet, Gérard. (1997). The zipper. *Journal of functional programming*, **7**(5), 549–554.

Jander, Piotr. (2019). *Verifying type-and-scope safe program transformations*. M.Phil. thesis, University of Edinburgh.

Jeffrey, Alan. (2011). *Associativity for free!* `http://thread.gmane.org/gmane.comp.lang.agda/3259`.

Kaiser, Jonas, Schäfer, Steven, & Stark, Kathrin. (2018). Binder aware recursion over well-scoped de bruijn syntax. *Pages 293–306 of: Proceedings of the 7th ACM SIGPLAN International Conference on Certified Programs and Proofs*. CPP 2018. ACM.

Keep, Andrew W., & Dybvig, R. Kent. (2013). A nanopass framework for commercial compiler development. *Sigplan not.*, **48**(9), 343–350.

Keuchel, Steven. (2011). *Generic programming with binders and scope*. M.Phil. thesis, Utrecht University.

Keuchel, Steven, & Jeuring, Johan. (2012). Generic conversions of abstract syntax representations. *Pages 57–68 of:* Löh, Andres, & Garcia, Ronald (eds), *Proceedings of the 8th ACM SIGPLAN workshop on Generic programming, WGP@ICFP 2012, Copenhagen, Denmark, September 9-15, 2012*. ACM.

Keuchel, Steven, Weirich, Stephanie, & Schrijvers, Tom. (2016). Needle & knot: Binder boilerplate tied up. *Pages 419–445 of: Proceedings of the 25th European Symposium on Programming Languages and Systems - Volume 9632*. Springer-Verlag New York, Inc.

Lee, Gyesik, Oliveira, Bruno C. D. S., Cho, Sungkeun, & Yi, Kwangkeun. (2012). Gmeta: A generic formal metatheory framework for first-order representations. *Pages 436–455 of:* Seidl, Helmut (ed), *Programming Languages and Systems*. Springer.

Licata, Daniel R., Shulman, Michael, & Riley, Mitchell. (2017). A fibrational framework for substructural and modal logics. *Pages 25:1–25:22 of:* Miller, Dale (ed), *2nd International Conference on Formal Structures for Computation and Deduction, FSCD 2017, September 3-9, 2017, Oxford, UK*. LIPIcs, vol. 84. Schloss Dagstuhl - Leibniz-Zentrum fuer Informatik.

Löh, Andres, & Magalhães, José Pedro. (2011). Generic programming with indexed functors. *Pages 1–12 of:* Järvi, Jaakko, & Mu, Shin-Cheng (eds), *Proceedings of the seventh ACM SIGPLAN workshop on Generic programming, WGP@ICFP 2011, Tokyo, Japan, September 19-21, 2011*. ACM.

Magalhães, José Pedro, Dijkstra, Atze, Jeuring, Johan, & Löh, Andres. (2010). A generic deriving mechanism for haskell. *Pages 37–48 of:* Gibbons, Jeremy (ed), *Proceedings of the 3rd ACM SIGPLAN Symposium on Haskell, Haskell 2010, Baltimore, MD, USA, 30 September 2010*. ACM.

Malcolm, Grant. (1990). Data structures and program transformation. *Sci. comput. program.*, **14**(2-3), 255–279.

Martin-Löf, Per. (1982). Constructive mathematics and computer programming. *Studies in logic and the foundations of mathematics*, **104**, 153–175.

The Coq Development Team. (2017). *The coq proof assistant reference manual*. πr^2 Team. Version 8.6.

McBride, Conor. (2005). *Type-preserving renaming and substitution*. Unpublished draft.

McBride, Conor. (2017). *Ornamental algebras, algebraic ornaments*. Unpublished draft.

McBride, Conor, & McKinna, James. (2004). The view from the left. *J. funct. program.*, **14**(1), 69–111.

McBride, Conor, & Paterson, Ross. (2008). Applicative programming with effects. *Journal of functional programming*, **18**(1), 1–13.

McLaughlin, Craig, McKinna, James, & Stark, Ian. (2018). Triangulating context lemmas. *Pages 102–114 of: Proceedings of the 7th ACM SIGPLAN Conference on Certified Programs and Proofs*. CPP 2018. ACM.

Mitchell, John C, & Moggi, Eugenio. (1991). Kripke-style models for typed lambda calculus. *Annals of pure and applied logic*, **51**(1-2), 99–124.

Morris, Peter, Altenkirch, Thorsten, & McBride, Conor. (2006). Exploring the regular tree types. *Pages 252–267 of:* Filliâtre, Jean-Christophe, Paulin-Mohring, Christine, & Werner, Benjamin (eds), *Types for Proofs and Programs*. Springer.

Neil Ghani, Makoto Hamana, Tarmo Uustalu, & Vene, Varmo. (2006). Representing cyclic structures as nested datatypes. *Pages 173–188 of: Proceedings of 7th Trends in Functional Programming, 2006*. Intellect.

Norell, Ulf. (2009). Dependently typed programming in Agda. *Pages 230–266 of: AFP Summer School*. Springer.

O'Hearn, Peter W. (2003). On bunched typing. *J. funct. program.*, **13**(4), 747–796.

Petricek, Tomas, Orchard, Dominic A., & Mycroft, Alan. (2014). Coeffects: a calculus of context-dependent computation. *Pages 123–135 of:* Jeuring, Johan, & Chakravarty, Manuel M. T. (eds), *Proceedings of the 19th ACM SIGPLAN international conference on Functional programming, Gothenburg, Sweden, September 1-3, 2014*. ACM.

Pickering, Matthew, Érdi, Gergő, Peyton Jones, Simon, & Eisenberg, Richard A. (2016). Pattern synonyms. *Pages 80–91 of: Proceedings of the 9th International Symposium on Haskell*. Haskell 2016. ACM.

Pierce, Benjamin C, & Turner, David N. (2000). Local type inference. *Acm transactions on programming languages and systems (toplas)*, **22**(1), 1–44.

Polonowski, Emmanuel. (2013). Automatically generated infrastructure for de bruijn syntaxes. *Pages 402–417 of:* Blazy, Sandrine, Paulin-Mohring, Christine, & Pichardie, David (eds), *Interactive Theorem Proving*. Springer.

Stump, Aaron. (2016). *Verified functional programming in Agda*. New York, NY, USA: Association for Computing Machinery and Morgan & Claypool.

Swiestra, Wouter. (2008). Data types à la carte. *Journal of functional programming*, **18**(4), 423–436.

Thibodeau, David, Momigliano, Alberto, & Pientka, Brigitte. (2016). *A case-study in programming coinductive proofs: Howe's method*. Tech. rept. Technical report, McGill University.

Wadler, Philip. (1987). Views: A way for pattern matching to cohabit with data abstraction. *Pages 307–313 of: Conference Record of the Fourteenth Annual ACM Symposium on Principles of Programming Languages, Munich, Germany, January 21-23, 1987*. ACM Press.

Wadler, Philip. (1990). Deforestation: Transforming programs to eliminate trees. *Theor. comput. sci.*, **73**(2), 231–248.

Wadler, Philip, & Kokke, Wen. (2018). *Programming language foundations in Agda*. Available at `http://plfa.inf.ed.ac.uk`.

Weirich, Stephanie, Yorgey, Brent A., & Sheard, Tim. (2011). Binders unbound. *Pages 333–345 of:* Chakravarty, Manuel M. T., Hu, Zhenjiang, & Danvy, Olivier (eds), *Proceeding of the 16th ACM SIGPLAN international conference on Functional Programming, ICFP 2011, Tokyo, Japan, September 19-21, 2011*. ACM.